

# **Canon**

# **X-720**

**Manuel d'utilisation  
du Module d'extension**

Le X-720 est un module d'extension destiné à l'ordinateur individuel CANON X-07. Le X-07 vient s'encastrier à l'avant du X-720.

Le X-720 permet à l'utilisateur de contrôler un téléviseur aux normes françaises ou un moniteur équipé d'une prise Périscopie (ou sans prise Périscopie avec un adaptateur antenne) à partir du X-07. Lorsque le X-07 est connecté à ce module, les fonctions du BASIC X-07 sont également étendues, et diverses opérations graphiques peuvent être effectuées sur l'écran du moniteur ou du téléviseur; d'autre part, le son du X-07 est reproduit par un téléviseur ou un moniteur si celui-ci est équipé d'une entrée son.

Enfin, le X-720 incorpore un logement de module d'extension enfichable. Le présent manuel décrit le fonctionnement du X-720 et la programmation avec le BASIC X-07 étendu. La première partie "FONCTIONNEMENT" est consacrée aux opérations élémentaires à effectuer avec le X-07, et la deuxième "PROGRAMMATION" à la programmation avec les extensions du langage BASIC X-07 permettant de réaliser l'affichage vidéo.

Nous vous conseillons de lire ce manuel avec soin afin d'exploiter au mieux toutes les possibilités offertes par le X-07 combiné au X-720.

#### **REMARQUE:**

CANON décline toute responsabilité au sujet de tout dommage, de tout préjudice ou de toute autre réclamation consécutifs à l'utilisation des programmes du présent manuel.

# TABLE DES MATIERES

## 1 ERE PARTIE: FONCTIONNEMENT


1	NOMENCLATURE ET FONCTION DES PIECES .....	1
2	RACCORDEMENT A UN TELEVISEUR .....	4
3	CONNEXION AU X-07 .....	6
4	RECHARGE DE LA BATTERIE .....	7
5	REPLACEMENT DE LA BATTERIE .....	8
6	REGLAGE .....	9
7	FICHE TECHNIQUE .....	10

## 2 EME PARTIE: PROGRAMMATION

1	PRESENTATION DU BASIC X-07 ETENDU .....	11
1-1	PAGE .....	12
1-2	MODE .....	13
1-3	COORDONNEES .....	14
1-4	COULEUR .....	17
2	ACCELERATION DES TRAITEMENTS GRAPHIQUES ..	18
3	INSTRUCTIONS ET FONCTIONS .....	19
3-1	INSTRUCTIONS .....	19
	SCREEN .....	19
	COLOR .....	24
	PSET, PRESET .....	33
	LINE .....	35
	CIRCLE .....	37
	PAINT .....	40
3-2	FONCTIONS .....	43
	POINT .....	43
	SCREEN .....	44
4	CHANGEMENT DE JEU DE CARACTERES .....	49
	ANNEXE .....	52




# ***PREMIERE PARTIE*** ***FONCTIONNEMENT***




# 1 Nomenclature et fonction des pièces

## Vue avant



## Vue arrière



- ① **Connecteur:** Ce connecteur se branche sur le connecteur de lignes de bus du X-07.
- ② **Prise d'adaptateur de secteur:** La fiche du câble de l'adaptateur de secteur (AD-5) se branche sur cette prise.
- ③ **Commande de remise à zéro:** Cette commande est réalisée à la commande correspondante du X-07.
- ④ **Crochet de fixation du X-07:** Ce crochet vient s'encaster dans l'encoche correspondante du X-07 de manière à le fixer sur le X-720.
- ⑤ **Bouton de verrouillage du X-07:** Lorsque ce bouton est en position haute, le X-07 est verrouillé sur le X-720. Pour libérer le X-07, actionner ce bouton vers le bas.
- ⑥ **Logement de module d'extension enfichable:** Ce logement est destiné à recevoir divers modules d'extension enfichables.
- ⑦ **Prise mâle Péritelévision: (sortie image et son)** Cette prise est destinée à être branchée sur la prise Péritelévision femelle:
- d'un téléviseur aux normes françaises;
  - d'un moniteur équipé d'une prise Péritelévision (avec entrée RVB-SYNCHRO-SON);
  - d'un adaptateur antenne (Ex.: C.G.V. PHS 60) pour raccordement à l'antenne d'un téléviseur SECAM.

## 2. Raccordement à un téléviseur:

### A. Téléviseur équipé d'une prise Périscopie:

Brancher la prise mâle Périscopie (réf. 7 sur croquis en P.1) dans la prise femelle correspondante du Téléviseur. En mettant le X-720 en service l'image apparaîtra automatiquement sur l'écran, indépendamment du canal de réception enclenché. En mettant le X-720 hors service, les émissions nationales réapparaîtront automatiquement à l'écran.


### B. Téléviseur non équipé d'une prise Périscopie:

Utiliser un adaptateur antenne (ex.: C.G.V. PHS 60).

## 3. Connexion au X-07

Connecter et déconnecter le X-07 comme indiqué ci-dessous.

**Avant de connecter ou de déconnecter le X-07, toujours couper son alimentation.**



- Connecter le X-07 au X-720 comme indiqué sur la figure ci-contre. Lorsque les deux ensembles sont correctement connectés, la languette du bouton de verrouillage du X-720 doit s'enclencher avec un déclic dans l'encoche correspondante du X-07.


- Actionner vers le bas le bouton de déverrouillage du X-720, et dégager le X-07 en le tirant dans le sens indiqué par la flèche sur la figure ci-contre.



## 4. Recharge de la Batterie

Si le niveau de tension de la batterie NiCd interne est insuffisant, le bon fonctionnement du X-720 ne peut pas être assuré, et l'affichage de l'image s'en ressent. Dans ce cas là, il est nécessaire de recharger la batterie pendant au minimum une heure. Utiliser pour ce faire l'adaptateur de secteur CANON AD-5 vendu séparément (ou fourni avec l'imprimante X-710). Lorsque la batterie a été rechargée pendant un certain temps, le X-720 peut être exploité avec la batterie en charge.


**Remarque:** Si le X-720 ne doit pas être utilisé pendant une longue période (24 heures ou plus), débrancher l'adaptateur AD-5. En effet, un dépassement de la durée normale de charge provoque une détérioration des performances de la batterie NiCd. Les performances de la batterie sont également détériorées lorsqu'elle demeure trop longtemps déchargée.




## 5. Remplacement de la batterie

Si en dépit d'une recharge d'une durée suffisante, le niveau de tension de la batterie NiCd chute au bout de quelques minutes, remplacer la batterie.

Utiliser une batterie NiCd du type spécifié par CANON et effectuer le remplacement comme indiqué ci-dessous.



1. Retourner le X-720. Retirer la vis de fixation du cache de compartiment de batterie, et déposer le cache.



2. Sortir le bloc de batterie NiCd du compartiment et tirer sur le connecteur de batterie comme indiqué sur la figure ci-contre.
3. Brancher au même endroit le connecteur de la nouvelle batterie, et rentrer la batterie dans le compartiment.
4. Remettre le cache en place, et le fixer à l'aide de la vis.

## 6. Reglage

Le programme vous permet de vérifier le bon fonctionnement de l'ensemble. Il affiche sur l'écran des bandes de couleur verticales, comme indiqué ci-contre.

Pour mettre fin au programme appuyer sur la touche **BREAK**.

Blanc
Violet
Jaune
Bleu
Noir
Cyan
Rouge
Vert

```
10 SCREEN 2,2,2
20 FOR X=0 TO 255 STEP 32
30 READ C
40 LINE (X,0)-(X+31,191),C,BF
50 NEXT X
60 GOTO 60
70 DATA 1,4,6,0,3,2,7,5
```

## 7. Fiche technique


<b>Type:</b>	Module d'extension destiné à la commande d'une unité de visualisation (utilisation avec le X-07 exclusivement).
<b>Entrée:</b>	Lignes de bus du X-07
<b>Sortie:</b>	Sortie de signal RVB-SYNCHRO-SON
<b>Fonctions:</b>	Extension du langage BASIC X-07 (affichage des couleurs, fonctions graphiques, etc.) Logement de module d'extension enfichable.
<b>Alimentation électrique:</b>	Batterie NiCd de 6 volts courant continu et adaptateur de secteur AD-5.
<b>Puissance consommée:</b>	1,4 W.
<b>Température ambiante d'utilisation:</b>	0°C à 40°C.
<b>Dimensions:</b>	220mm (l.) × 270mm (pr.) × 53,5mm (ht.)
<b>Poids:</b>	1,3 K.

Sous réserve de modifications sans préavis



# ***DEUXIEME PARTIE***

## ***PROGRAMMATION***



# 1. Présentation du BASIC X-07 étendu

Puisque le X-720 permet d'utiliser non seulement l'affichage incorporé au X-07, mais également un affichage sur écran de téléviseur ou de moniteur, le langage BASIC X-07 a dû être étendu. Le BASIC X-07 étendu est constitué par le BASIC X-07 utilisé avec le X-07 seul, avec des fonctions étendues pour un certain nombre d'instructions préexistantes, et avec un certain nombre d'instructions nouvelles.

Les différences entre le simple BASIC X-07 et le BASIC X-07 étendu sont les suivantes:

- **Les instructions suivantes présentent des fonctions étendues:**

- PSET, PRESET**

- Désignation de couleurs rajoutée

- LINE** ..... Désignation de couleurs et tracé d'un rectangle rajoutés

- CIRCLE** ..... Désignation de couleurs et tracé d'un cercle rajoutés

- fonction

- POINT** ..... Confirmation d'une couleur sur l'écran

- fonction

- SCREEN** ..... Lecture des informations d'image

- **Nouvelles instructions et nouvelles fonctions**

- SCREEN** ..... Spécification de mode écran


- COLOR** ..... Spécification de couleurs

- PAINT** ..... Coloration d'une zone

Les termes utilisés pour décrire ces fonctions étendues sont définis ci-après.

## 1-1 Page

Le X-720 permet d'utiliser non seulement l'affichage à cristaux liquides (LCD du X-07 mais également l'écran d'un téléviseur ou d'un moniteur. Avec un écran de téléviseur ou de moniteur, on peut utiliser deux types d'affichage, en fonction des modes d'écrans décrits ultérieurement. Finalement, on distingue trois types d'écrans appelés "pages", qui sont utilisés en fonction du but recherché. Ces pages sont numérotées de 1 à 3 comme indiqué ci-dessous.



**Page 1:** LCD du X-07

**Page 2:** Ecran définition standard

**Page 3:** Ecran haute définition

La manière dont ces différentes pages sont utilisées est indiquée en détail à la rubrique "SCREEN" du présent manuel. Les pages portent également différents noms suivant leur fonction.

Les pages sur lesquelles une instruction PRINT ou diverses instructions graphiques doivent être exécutées, c'est-à-dire les pages sur lesquelles des caractères ou des graphiques doivent être tracés, constituent ce que l'on appelle des pages actives, et l'écran en cours d'affichage (celui qui est visible) constitue ce que l'on appelle la page visuelle.



## 1-2 Mode

En fonction du but recherché, le mode de fonctionnement de l'écran peut être sélectionné parmi six modes différents.

Ces modes présentent les caractéristiques indiquées ci-dessous, et sont sélectionnés à l'aide de l'instruction SCREEN.

### **Mode 1 ... Mode texte**

Ce mode est utilisé pour l'entrée et la sortie de caractères ordinaires.

### **Mode 2 ... Mode semi-graphique**

Ce mode permet l'entrée et la sortie de caractères ordinaires et les traitements graphiques en  $64 \times 48$  points/page.

### **Mode 3 ... Mode couleur définition standard**

Ce mode permet d'obtenir l'écran graphique de  $128 \times 96$  points/page. Quatre couleurs peuvent être simultanément spécifiées en unités de point.

### **Mode 4 ... Mode monochrome définition standard**

Ce mode permet d'obtenir l'écran graphique de  $128 \times 192$  points/page. Un affichage monochrome noir et blanc ou vert et noir peut être spécifié en unités de point.

### **Mode 5 ... Mode couleur haute définition**

Ce mode permet d'obtenir l'écran graphique de  $128 - 192$  points/page. Quatre couleur peuvent être spécifiées simultanément en unité de point.

### **Mode 6 ... Mode monochrome haute définition**

Ce mode permet d'obtenir l'écran graphique de  $256 \times 192$  points/page. L'affichage monochrome noir et blanc ou vert et noir peut être spécifié en unités de point.

Chacun de ces modes est décrit en détail à la rubrique consacrée à l'instruction SCREEN dans le présent manuel.




### 1-3 Coordonnées

Bien que la définition de l'affichage sur l'écran (c'est-à-dire le nombre de points horizontaux et verticaux) varie en fonction du mode sélectionné comme précédemment indiqué, le BASIC X-07 étendu n'utilise qu'un seul système de coordonnées pour effectuer un repérage ponctuel sur des écrans présentant diverses définitions.

Ces coordonnées sont appelées coordonnées virtuelles.

Les chiffres de 0 à 255 sont affectés à l'axe horizontal de l'écran, et les chiffres de 0 à 191 à l'axe vertical, comme indiqué sur la figure ci-dessous.




La manière dont cet unique système de coordonnées virtuelles s'applique à des écrans présentant diverses définitions est expliquée ci-dessous.

Comparons par exemple une page en mode 2 (définition de  $64 \times 48$  points) et en mode 6 (définition de  $256 \times 192$  points). Bien qu'en mode 2 la page ne comporte que 64 points horizontaux et 48 points verticaux, les chiffres de 0 à 255 n'en sont pas moins affectés à l'axe horizontal, et ceux de 0 à 191 à l'axe vertical.


Dans ce cas-là, les coordonnées correspondant aux différents points de la page sont indiquées ci-dessous.

Système de correspondance des coordonnées virtuelles en mode 2.




Cela signifie que, en mode 2, les coordonnées (0,0) et (3,3) spécifient le même point. Par contre, en mode 6, la définition étant de 256 points horizontaux  $\times$  192 points verticaux, les coordonnées (0,0) et (3,3) spécifient deux points différents.

### Système de coordonnées en mode 6.




Le schéma ci-dessous illustre la correspondance des coordonnées entre les mode 2 et 6.



## 1-4 Couleur


Lorsque le X-720 est connecté à une unité de visualisation couleur, diverses couleurs peuvent être spécifiées pour différents traitements graphiques. La méthode de spécification des couleurs est décrite ultérieurement. Dans la présente rubrique quelques schémas vont nous permettre de définir les termes utilisés dans le cadre de la spécification des couleurs.

**Couleur d'avant-plan:** Couleur des caractères ou des points.



**Couleur d'arrière-plan:** Couleur d'effacement d'écran ou de l'arrière-plan.

**Couleur de la bordure:** Couleur de la bordure entourant la zone d'affichage.



Il faut noter que la définition des couleurs d'avant-plan et d'arrière-plan et que la méthode de spécification des couleurs diffèrent en fonction du mode écran utilisé.

## 2. Accelération des traitements graphiques

Divers traitements graphiques peuvent être accélérés en affectant à l'adresse &H4BA une valeur autre que 0. Cependant, ce procédé entraîne un changement d'écran.

Comparer la différence des vitesses de déroulement des deux programmes suivants:

```
10 SCREEN 5:POKE &H4BA,0
20 COLOR 1,3,1:CLS
30 CIRCLE (100,100),80
40 PAINT (100,100)
50 GOTO 50
```

```
10 SCREEN 5:POKE &H4BA,1
20 COLOR 1,3,1:CLS
30 CIRCLE (100,100),80
40 PAINT (100,100)
50 GOTO 50
```

## 3 Instructions et fonctions

### 3-1 Instructions

# SCREEN

Fonction	Cette instruction sert à spécifier le monde écran et les pages.
Format	SCREEN [<mode écran>], [<page active>], [<page visuelle>]
Exemple	SCREEN 2, 3, 3

**Explications** Cette instruction spécifie le mode écran, la page active et la page visuelle. <mode écran> spécifie l'un des numéros de mode de 1 à 6 <page active> et <page visuelle> spécifient un numéro de page de 1 à 3.

Les pages pouvant être spécifiées dans chacun des modes sont indiquées ci-dessous.

Page	1 (LCD du X-07)	2 (écran de l'unité de visualisation)	3 (écran de l'unité de visualisation)
Modes	1,2	1,2,3,4	1,2,5,6

Lorsque <mode écran> est omis, la valeur spécifiée par <page active> est automatiquement spécifiée. Chacun des modes est décrit ci-dessous.

### **Mode 1 (mode texte)**

- L'entrée/sortie de caractères ordinaires est effectuée dans ce mode.
- On peut afficher 32 caractères × 16 lignes.
- La définition est de 32 points horizontaux × 16 points verticaux.

#### **Exemple de programme:**

Effectuer un programme d'édition, etc. sur un écran d'unité de visualisation.

```
SCREEN 1,2
```

#### **Exemple de programme:**

Changement de page

```
10 SCREEN 1,2,3:CLS
20 PRINT "A"
30 SCREEN 1,3,2:CLS
40 PRINT " B"
50 SCREEN 1,2,2
60 FOR I=1 TO 50:NEXT I
70 SCREEN 1,3,3
80 FOR I=1 TO 50:NEXT I
90 GOTO 50
```

### **Mode 2 (mode semi-graphique)**

- On peut spécifier 8 couleurs plus le noir en unité de caractères.
- On peut afficher 32 caractères × 16 lignes.
- L'entrée/sortie de caractères ordinaires est également possible.

**Exemple de programme:**

```
10 SCREEN 2,2,2:CLS
20 FOR I=0 TO 200 STEP 40
30 FOR J=0 TO 200 STEP 40
40 LINE (I,0)-(J,191)
50 NEXT J,I
```

Remarque: Avec l'affichage à cristaux liquides du X-07, les modes 1 et 2 sont supposés être identiques.

**Mode 3 (mode couleur avec la définition standard)**

- Quatre couleurs peuvent être simultanément affichées en unités de points.
- On peut afficher 16 caractères × 8 lignes.
- Les caractères peuvent être réécrits.
- Ce mode peut être simultanément spécifié avec la page 3 en mode 1 ou 2.

**Exemple de programme:**

Réécriture de caractères

```
10 SCREEN 3,2,2:CLS
20 PRINT "ABC"
30 LOCATE 0,0
40 PRINT "XYZ"
50 GOTO 50
```

Remarque: Puisque les modes 3,4,5 et 6 sont des modes graphiques, à la fin de l'exécution du programme, ou dans l'attente d'une entrée à la suite d'une instruction INPUT, l'écran d'entrée/sortie abandonne le mode 3 ou 4 pour revenir à la page 3 en mode 1 ou 2, ou l'écran abandonne le mode 5 ou 6 pour revenir à l'affichage par cristaux liquides du X-07.



#### **Mode 4 (mode monochrome avec la définition standard)**

- Le noir et blanc ou le vert et le noir peuvent être affichés en unités de points.
- On peut afficher 16 caractères × 16 lignes.
- La définition est de 128 points horizontaux × 192 points verticaux.
- La réécriture de caractères est possible
- Ce mode peut être spécifié en même temps que la page 3 en mode 1 ou 2.

Remarque: Lorsque le mode 3 ou 4 est spécifié, la page 2 devient automatiquement la page active quelle que soit la spécification de <page active>.

#### **Mode 5 (mode couleur avec la haute définition)**

- Quatre couleurs peuvent être simultanément affichées en unité de points.
- On peut afficher 16 caractères × 16 lignes.
- La définition est de 128 points horizontaux × 192 points verticaux.
- La réécriture de caractères est possible.

Remarque: En modes 5 et 6, la page 3 constitue la page active et la page visuelle, quels que soient les paramètres spécifiés. La page 2 ne peut être utilisée avec aucun mode.

#### **Exemple de programme:**

Exécution de l'exemple relatif au mode 2 en mode 5

```
10 SCREEN 5:CLS
20 FOR I=0 TO 200 STEP 40
30 FOR J=0 TO 200 STEP 40
40 LINE (I,0)-(J,191)
50 NEXT J,I
60 GOTO 60
```



### Mode 6 (mode monochrome avec la haute définition)

- Le noir et blanc ou le noir et vert peut être affiché en unités de points.
- On peut afficher 32 caractères × 16 lignes.
- La définition est de 256 points horizontaux × 192 points verticaux.

#### Exemple de

#### Programme: Tracé d'un cercle

```
10 SCREEN 6:CLS
20 FOR I=10 TO 80 STEP 10
30 CIRCLE (100,100),I
40 NEXT I
50 GOTO 50
```

Le tableau ci-dessous récapitule la définition et le nombre de caractères à l'affichage pour chacun des modes.

Mode	Définition (pts)	Nb de caractères à l'affichage
1	32 × 16	32 caractères × 16 lignes
2	64 × 48	32 × 16 "
3	128 × 96	16 × 8 "
4	128 × 192	16 × 16 "
5	128 × 192	16 × 16 "
6	256 × 192	32 × 16 "

# COLOR

Fonction:	Cette instruction spécifie les couleurs à afficher.
Format:	COLOR [<couleur d'avant-plan>] [, [<couleur d'arrière-plan>] [, <jeu de couleurs>]]
Exemple:	COLOR 1, 3, 1

**Explications:** Cette instruction spécifie la couleur des caractères à afficher, des traitements graphiques, etc. Une fois que cette instruction a été exécutée, les couleurs spécifiées sont automatiquement utilisées pour réaliser un affichage lorsque la spécification des couleurs est omise.

<jeu de couleurs> spécifie la combinaison de couleurs à utiliser. Le numéro spécifié est soit 1 soit 2.

Lorsque l'on spécifie une valeur située au-delà de la limite supérieur de l'intervalle de définition, cette valeur est automatiquement remplacée par la valeur maximale autorisée. Lorsque l'on spécifie une valeur située en-deçà de la limite inférieure de l'intervalle de définition, cette valeur est automatiquement remplacée par la valeur minimale autorisée. Par exemple, si un mode ne permet de spécifier qu'un numéro de couleur de 1 à 4, si l'on spécifie 6, cette valeur est automatiquement remplacée par 4, et si l'on spécifie 0,5, cette valeur est automatiquement remplacée par 1. Cette règle s'applique également aux numéros de couleurs spécifiés dans le cadre d'instructions graphiques.

Si l'un des paramètres est omis, la précédente spécification de ce même paramètre est automatiquement adoptée.

La spécification des couleurs varie en fonction du mode utilisé; les diverses méthodes de spécification sont indiquées dans les pages suivantes.

Les numéros de couleurs utilisés dans le cadre de ces explications, sont les mêmes que ceux spécifiés dans le cadre des instructions CIRCLE, LINE, PSET, etc.

## Mode 1

N° de couleur	Jeu de couleurs N°1	Jeu de couleurs N°2
1	Vert	Rouge
2	Vert clair	Orange
3	Rouge	Vert
4	Orange	Vert clair

Les couleurs correspondant aux paramètres de caractère, d'arrièreplan de caractères et d'instruction PRESET sont indiquées ci-dessous. Les couleurs correspondant à CLS sont celles qui s'affichent sur l'écran lorsqu'il a été effacé par l'instruction CLS.

Couleur d'avant-plan	Jeu de couleurs N°1			Jeu de couleurs N°2		
	Caractère	Arrière-plan de caractère	CLS	Caractère	Arrière-plan de caractère	CLS
1	Vert clair	Vert	Vert	Orange	Rouge	Rouge
2	Vert	Vert clair	Vert clair	Rouge	Orange	Orange
3	Orange	Rouge	Rouge	Vert clair	Vert	Vert
4	Rouge	Orange	Orange	Vert	Vert clair	Vert clair

PRESET permet de spécifier la couleur de l'arrière-plan. Si on exécute un PRESET à un emplacement où un caractère est affiché, la couleur d'affichage change.

Couleur d'avant-plan	Jeu de couleurs N°1			Jeu de couleurs N°2		
	PRESET	Caractère	Arrière-plan de caractère	PRESET	Caractère	Arrière-plan de caractère
1	Vert	Vert clair	Vert	Rouge	Orange	Rouge
2	Vert clair	Vert	Vert clair	Orange	Rouge	Orange
3	Rouge	Orange	Rouge	Vert	Vert clair	Vert
4	Orange	Rouge	Orange	Vert clair	Vert	Vert clair

Exécuter le programme suivant tout en consultant le tableau.

```
10 SCREEN 1,2,2
20 INPUT "FOREGROUND";F
30 INPUT "BACKGROUND";B
40 COLOR F,B,1
50 PRINT "COLOR ";F;" ";B;" ";1
60 FOR I=120 TO 192 STEP 12
70 PRESET (0,I)
80 NEXT I
90 GOTO 20
```

La bande verticale tracée sur l'écran à partir du coin supérieur gauche est affichée par PRESET.

```
10 SCREEN 1,2,2
20 COLOR 1,2,1:CLS
30 CIRCLE (100,100),80
40 GOTO 40
```

Un cercle doit être tracé, mais rien n'apparaît sur l'écran. En effet, une instruction CIRCLE sans spécification couleur, trace un cercle dont la couleur est identique à celle d'effacement d'écran. Si on spécifie une couleur d'arrière-plan pour l'écran, à l'aide d'une instruction PRESET comme dans le programme ci-dessous, on peut distinguer le cercle.


```
10 SCREEN 1,2,2
20 COLOR 1,2,1:CLS
21 FOR I=0 TO 255 STEP 8
22 FOR J=0 TO 191 STEP 12
23 PRESET (I,J)
24 NEXT J,I
30 CIRCLE (100,100),80
40 GOTO 40
```

## Mode 2

N° de couleur	Jeu de couleurs N° 1	Jeu de couleurs N° 2
0	Noir	Noir
1	Vert	Blanc
2	Jaune	Cyan
3	Bleu	Violet
4	Rouge	Orange
5	Blanc	Vert
6	Cyan	Jaune
7	Violet	Bleu
8	Orange	Rouge

Couleur de la bordure: Noir

La couleur d'avant-plan et la couleur d'arrière-plan sont spécifiées par les numéros de couleur énumérés ci-dessus. La relation entre la couleur d'affichage et la couleur spécifiée est indiquée ci-dessous.



Cependant, lorsque les caractères sont affichés en mode 2, la spécification de <couleur d'avant-plan> a la même signification qu'en mode 1. Autrement dit, l'affichage de caractères se fait de la même manière en mode 1 et en mode 2. Par conséquent, même si la couleur d'avant-plan est spécifiée à l'aide d'un chiffre égal ou supérieur à 5, ce chiffre est automatiquement remplacé par 4 pour l'affichage de caractères, de même qu'en mode 1, et les numéros de couleurs du mode 1 s'appliquent au mode 2.

Exécuter le programme suivant et vérifier la correspondance entre l'affichage des caractères et l'affichage graphique.

```
10 SCREEN 2,2,2
20 INPUT "FOREGROUND" ;F
30 INPUT "BACKGROUND" ;B
40 COLOR F,B,1:CLS
50 PRINT "COLOR " ;F ;", " ;B ;", 1"
60 LINE (0,0)-(100,100)
70 END
```

Lors de l'exécution du programme ci-dessus, entrer 0 après l'affichage de "FOREGROUND?" ou de "BACKGROUND?". L'irrégularité du trait tracé à partir du coin supérieur gauche jusqu'au coin inférieur droit de l'écran diffère de celle obtenue lorsque l'on spécifie un autre numéro de couleur. Ce point correspond à la définition du mode 2 original.

En mode 2, la couleur peut changer dans l'espace occupé par un caractère sur l'écran. Cependant, deux couleurs, ou un nombre supérieur de couleurs, ne peuvent pas être affichées dans l'espace occupé par un caractère. Par conséquent, des couleurs peuvent être mélangées dans l'espace occupé par un caractère, et un trait et un cercle peuvent être tracés avec la définition originale uniquement lorsque le noir, qui n'est pas considéré comme une couleur, est combiné à une des couleurs.

Vérifier ce point en exécutant le programme suivant:



```

10 SCREEN 2,2,2
20 COLOR 3,4,1:CLS
30 CIRCLE (100,100),80
40 SCREEN 2,3,3
50 COLOR 3,0,1:CLS
60 CIRCLE (100,100),80
70 SCREEN 2,2,2
80 FOR I=1 TO 100:NEXT I
90 SCREEN 2,3,3
100 FOR I=1 TO 100:NEXT I
110 GOTO 70

```


### Mode 3

### Mode 5

En mode 3 et en mode 5 la spécification des couleurs est la même.

N° de couleur	Jeu de couleurs N° 1	Jeu de couleurs N° 2
1	Vert	Blanc
2	Jaune	Cyan
3	Bleu	Violet
4	Rouge	Orange
Couleur de bordure	Vert	Blanc

La correspondance entre la couleur d'affichage et la couleur spécifiée est indiquée ci-dessous.



**Exemple de programme:**

```

10 SCREEN 5
20 COLOR 1,4,1:CLS
30 FOR C=1 TO 3
40 FOR X=0 TO 255 STEP 32
50 COLOR C,4,1
60 LINE ((C-1)*127,0)-(X,191)
70 NEXT X,C
80 GOTO 80

```



## Mode 4

## Mode 6

Les spécifications de couleurs sont les mêmes en mode 4 et en mode 6. La correspondance entre la couleur d'affichage et la couleur spécifiée est la même que pour les modes 3 et 5. Cependant, il faut noter que les numéros de couleur des modes 4 et 6 diffèrent de ceux des autres modes du fait qu'ils commencent par zéro.

N° de couleur	Jeu de couleurs N°1	Jeu de couleurs N°2
0 1	Noir Vert	Noir Blanc
Couleur de Bordure	Vert	Blanc

Remarque: En fonction du type de l'unité de visualisation utilisé, il peut arriver que l'on obtienne une couleur avec la spécification noir/blanc du mode 6.

### Exemple de programme:

```
10 SCREEN 6
20 COLOR 1,0,1:CLS
30 A=0:B=1
40 FOR I=0 TO 255 STEP 12
50 LINE (128,A*191)-(I,B*191)
60 NEXT I
70 IF A=0 THEN A=1:B=0:GOTO 40
80 END
```

# PSET

## PRESET

**Fonction:** Cette instruction permet de tracer un point sur la page active.

**Format:**

PSET	[STEP] (<abscisse X>, <ordonnée Y>),
PRESET	

  
[, <N° de couleur>]

**Exemple:** PRESET (100, 100), 2

**Explications:** Cette instruction permet de tracer un point sur la page active à l'emplacement indiqué par les coordonnées. Les coordonnées spécifiées doivent respecter l'intervalle de définition des coordonnées d'écran ( $0 \leq X \leq 255$ ,  $0 \leq Y \leq 191$ ).

Les numéros définis pour chaque mode sont spécifiés par <N° de couleur>.

Lorsque <N° de couleur> est spécifié, PSET et PRESET ont la même fonction.

Lorsque <N° de couleur> est omis, la couleur du point tracé par PSET est celle d'avant-plan, et la couleur du point tracé par PRESET est celle d'arrière-plan. (pour plus de précisions, se reporter aux explications fournies pour chacun des modes).

Le jeu de couleurs spécifié par la précédente instruction COLOR demeure valide. (Si aucune instruction COLOR n'a été exécutée, le jeu de couleurs N° 1 est automatiquement spécifié.)

### Exemple de programme:

```
10 SCREEN 5
20 COLOR 2,3,1:CLS
30 FOR I=0 TO 255
40 A=60*SIN(I*(6.28/255))
50 PSET (I,A+100)
60 NEXT I
```

# LINE

**Fonction:** Cette instruction permet de tracer un trait ou un rectangle.

**Format:** LINE [STEP] [( $\langle$ abscisse X $\rangle$ ,  $\langle$ ordonnée Y $\rangle$ )] –  
[SETP] ( $\langle$ abscisse x $\rangle$ ,  $\langle$ ordonnée y $\rangle$ ) [ $\langle$ N° de  
couleur $\rangle$ ]  
[, | B | ]  
| BF |

**Exemple:** LINE (0, 0) – (100, 100),, B  
LINE – (60, 25), 3, BF

**Explications:** Cette instruction permet de relier un point de l'écran ( $\langle$ abscisse X $\rangle$ ,  $\langle$ ordonnée Y $\rangle$ ) à un autre ( $\langle$ abscisse x $\rangle$ ,  $\langle$ ordonnée y $\rangle$ ) par une ligne droite. Lorsque B est spécifié, un rectangle est tracé, les deux points spécifiés se retrouvant aux extrémités de la diagonale du rectangle. Les points spécifiés doivent respecter l'intervalle de définition des coordonnées d'écran ( $0 \leq X, x \leq 255$ ,  $0 \leq Y, y \leq 191$ ).

Lorsque BF est spécifié, l'intérieur du rectangle ainsi tracé est colorié.

La couleur du trait du rectangle est spécifiée par  $\langle$ N° de couleur $\rangle$ . Au sujet de la spécification de N° de couleur, se reporter aux explications relatives à l'instruction COLOR correspondant au mode utilisé.

Le jeu de couleurs spécifié par la précédente instruction COLOR demeure valide. (Si aucune instruction COLOR n'a été exécutée, le jeu N° 1 est automatiquement spécifié.)

**Exemple de programme:**

```
10 SCREEN 5
20 CLS:PI=3.14
30 X=255:Y=191/2
40 LINE (X,Y)-(0,Y),4
50 FOR I=1 TO X-1
60 LINE-(I,-Y*SIN(PI*I/20)+Y),4
70 NEXT I
80 GOTO 80
```

**Exemple de  
programme:**

```
10 SCREEN 2
20 COLOR 1,0,1:CLS
30 X=0
40 FOR I=1 TO 8
50 X=X+12
60 LINE (X,X)-(X+80,X+80),I,BF
70 NEXT I
80 GOTO 80
```


# CIRCLE

Fonction:	Cette instruction permet de tracer un cercle ou une ellipse.
Format:	CIRCLE [SETP] (<abscisse X>, <ordonnée Y>), <rayon> [, [<N° de couleur>] [, <coefficient de forme>]]
Exemple:	CIRCLE (100, 100), 50, 2

**Explications:** Cette instruction permet de tracer, en fonction du rayon spécifié, un cercle autour du point dont les coordonnées sont spécifiées. La spécification <N° de couleur> est identique à celle utilisée dans le cadre de l'instruction LINE.  
<coefficient de forme> spécifie un nombre entier de 0 à  $\pm 256$ . Ce coefficient indique le taux de compression du cercle, et permet ainsi de tracer une ellipse. Il faut noter que l'on obtient un message d'erreur lorsque les spécifications ne permettent pas de tracer un cercle ou une ellipse entièrement contenu dans les limites de l'écran.  
Comme indiqué ci-dessous, le profil de l'ellipse est fonction du coefficient de forme.


- Lorsque le coefficient de forme est  $\geq 0$  ...  
l'ordonnée X des points situés sur la circonférence est comprimée par le coefficient de forme/256

Coefficient de forme = 100



- Lorsque le coefficient de forme est  $< 0$  ...  
l'abscisse X des points situés sur la circonférence est comprimée par le coefficient de forme/256

Coefficient de forme = -100



Si le coefficient de forme est omis, ou si  $\pm 256$  est spécifié, on obtient un cercle.

**Exemple de programme:**

```

10 SCREEN 5
20 COLOR 1,3,1:CLS
30 S=32:C=32
40 PRINT "S=";S
50 CIRCLE (128,96),90,,S
60 CLS
70 S=S+C
80 IF S=-256 THEN S=256:GOTO 40
90 IF S=256 THEN S=-256:GOTO 40
100 IF S=0 THEN C=C*-1:GOTO 40
110 IF S=-32 THEN C=C*-1:GOTO 40
120 GOTO 40

```

Remarque: En fonction du réglage de l'unité de visualisation et de la structure de points, le cercle tracé sur l'écran peut ne pas être parfait.

# PAINT

Fonction:	Cette instruction permet de colorier une surface fermée affichée sur l'écran.
Format:	PAINT [STEP] (<abscisse X>, <ordonnée Y>) [, [<N° de couleur>] [, <couleur du pourtour>]]
Exemple:	PAINT (100, 100), 2 PAINT STEP (10,0), 2, 3

**Explications:** Cette instruction permet de colorier, sur l'écran, avec la couleur spécifiée la surface fermée contenant le point spécifié. La surface fermée correspond à la zone délimitée par des traits ou des figures tracés dans la couleur de pourtour spécifiée et le bord de l'écran. Si la partie à colorier est située dans cette zone, le résultat est le même quel que soit le point spécifié. Si <N° de couleur> ou <couleur de pourtour> est omis, la couleur d'avant-plan est automatiquement spécifiée. Le jeu de couleurs spécifié par la précédente instruction COLOR demeure valide. (Si aucune instruction COLOR n'a été exécutée, le jeu N° 1 est automatiquement spécifié.)

Cette instruction **n'est pas valide en mode 2.**

Pour Vérifier la méthode de coloration avec l'instruction PAINT, exécuter le programme suivant:

```
10 SCREEN 3,2,2
20 COLOR 1,4,1:CLS
30 LINE (0,0)-(191,191)
40 PAINT (100,0),3
50 GOTO 50
```

Lorsque ce programme est exécuté, la moitié droite de l'écran est coloriée par une ligne droite partant du point supérieur gauche pour aboutir au bas de l'écran. Maintenant, rajouter au programme précédent la ligne 31 ci-dessous, et exécuter de nouveau le programme.

```
31 PRESET (100,100)
```

Puisqu'une partie de la ligne droite est coupée par l'instruction PRESET de la ligne 31, la moitié droite de l'écran n'est plus une surface fermée et la totalité de l'écran en dehors de la ligne droite est coloriée.



Lorsque l'on utilise l'instruction PAINT de cette manière il faut faire extrêmement attention pour savoir quel trait délimite la zone à colorier. Avant d'aborder les explications relatives à la couleur du pourtour, modifier le programme comme indiqué ci-dessous et l'exécuter une nouvelle fois.

```
10 SCREEN 3,2,2
20 COLOR 1,4,1:CLS
30 LINE (0,0)-(191,191)
31 LINE (0,191)-(191,0),3
40 PAINT (100,0),3,3
50 GOTO 50
```

Avec le programme ainsi modifié, la moitié gauche de l'écran est colorié par une droite partant du coin inférieur gauche pour aboutir au coin supérieur droit avec la nouvelle couleur N° 3 pour le pourtour.

Puisque dans le programme initial une instruction PAINT avait été exécutée en omettant le paramètre <couleur de pourtour>, la ligne droite tracée dans la couleur d'avant-plan (couleur N° 1) a été prise comme limite de zone.

Dans le programme modifié, par contre, la couleur N° 3 est spécifiée comme couleur de pourtour. Par conséquent, la ligne tracée dans la couleur N° 1 est ignorée comme limite de zone et la ligne tracée dans la couleur N° 3 est prise comme pourtour.

Lorsque l'on utilise l'instruction PAINT de cette manière, il faut également faire extrêmement attention à la couleur du pourtour.

**Exemple de  
Programme:**

```
10 SCREEN 5
20 COLOR 2,1,1:CLS
30 CIRCLE (125,60),50
40 CIRCLE (150,110),50
50 CIRCLE (100,110),50
60 PAINT (125,40)
70 PAINT (155,110),4
80 PAINT (90,110),3
90 PAINT (150,80),3
100 PAINT (100,80),4
110 PAINT (125,140),2
120 GOTO 120
```



## 3-2 Fonctions

# POINT

<b>Fonction:</b>	Cette fonction fournit l'information de couleur correspondant au point spécifié.
<b>Format:</b>	POINT (<abscisse X>, <ordonnée Y>)
<b>Exemple:</b>	A=POINT (100, 100)

**Explications:** Cette fonction fournit le N° de la couleur du point spécifié par les coordonnées. Les numéros de couleur sont ceux correspondant à chacun des modes.

En mode 1, lorsque des caractères sont affichés sur un arrière-plan, si les coordonnées spécifiées correspondent à un emplacement occupé par un caractère, cette fonction fournit le numéro de la couleur de l'arrière-plan du caractère.


Par contre, en mode 2, si les coordonnées spécifiées correspondent à un emplacement de caractère, cette fonction n'a aucun sens. Si les coordonnées spécifiées aboutissent à un point situé en dehors de l'écran, on obtient la réponse -1.

# SCREEN

Fonction:	Cette fonction fournit l'information d'image affichée sur l'écran.
Format:	SCREEN (<abscisse x de l'image>, <ordonnée y de l'image>)
Exemple:	PRINT SCREEN (0,0)

**Explications:** Cette fonction fournit les informations correspondant à l'image ou aux caractères; ces informations sont affichées sur l'écran sous forme numérique. Le mode information d'image diffère suivant le mode écran. La position d'image X,Y correspond aux coordonnées d'écran fixées pour la lecture d'informations d'image en fonction de chacun des modes. Il faut bien noter que ces coordonnées diffèrent de celles utilisées pour le tracé des figures, etc. Si les coordonnées ne respectent pas l'intervalle de définition de position d'image fixé pour le mode concerné, on obtient un message d'erreur.

## Mode 1




En mode 1, cette fonction fournit le code du caractère à afficher à la position spécifiée. L'intervalle de définition de position d'image est défini de la manière suivante:  $0 \leq X \leq 31$ ,  $0 \leq Y \leq 15$  en unités de caractère comme indiqué dans l'illustration de l'écran.


En mode 1, la valeur fournie par cette fonction n'a rien à voir avec la couleur affichée. Par exemple, lorsque "A" est affiché dans le coin supérieur gauche de l'écran, SCREEN (0,0) a la valeur 65.

En mode 2, cette fonction peut fournir le code caractère comme en mode 1, ou aucune information. L'intervalle de définition de position d'image est défini de la manière suivante:  $0 \leq X \leq 31$ ,  $0 \leq Y \leq 15$ .


## Mode 2



Pour les zones de l'écran où des caractères sont affichés, cette fonction opère exactement comme dans le mode 1. Lorsque l'on utilise une instruction graphique pour effectuer un tracé à l'aide de points sur un arrière-plan noir, puisque six points occupent dans ce cas là un emplacement de caractère, la valeur fournie par cette fonction a la signification suivante:





Tracer par exemple les points indiqués ci-dessous dans la couleur N° 3 sur un fond noir à la position d'image (31,0) dans le coin supérieur droit de l'écran.




Puisque les points correspondant aux bits 2, 8, 32 ont été tracés, l'information de points correspond à  $2+8+32=58$ . De plus, puisque l'information de couleur correspond aux 128 bits nécessaires pour représenter la couleur N° 3 ou 7, la valeur de SCREEN (31, 0) est = à  $58 + 128 = 186$

### Mode 3 ou 5




En modes 3 et 5, la position d'image est fixée par unités de quatre points. Par conséquent, l'information sur l'écran peut être lue par unité de quatre points.

L'information de points est convertie en code numérique d'information de couleur pour chacun des deux points comme indiqué ci-dessous.





Tracer par exemple à l'aide des quatre points de la position d'image (10,0) de l'écran ayant la structure indiqué ci-dessous en mode 5.




L'information d'image correspondant à la position d'image (10,0) présente la structure de bit suivante: "11100000". SCREEN (10,0) a comme valeur  $128 + 64 + 32 = 224$ .

### Mode 4 ou 6



En modes 4 et 6, la position d'image est fixée par unités de 8 points. par conséquent, l'information de points de l'écran peut être lue par unité de 8 points.

8 bits sont affectés à chacun des points de l'information de points. Suivant que le point doit avoir la couleur N° 1 ou la couleur N° 0, le bit correspondant est à 1 ou à 0.



**Exemple de  
programme:**

Doubler, dans le sens de la largeur, la taille du caractère inscrit dans le coin supérieur gauche, et l'afficher au centre de l'écran.

```
10 SCREEN 4,2,2
20 COLOR 1,0,1:CLS
30 PRINT "A"
40 DIM D(8)
50 FOR I=0 TO 23
60 P=SCREEN(0,I)
70 FOR S=0 TO 7
80 D(S+1)=P AND 2^S
90 IF D(S+1)<>0 THEN D(S+1)=1
100 NEXT S
110 FOR M=1 TO 8
120 PSET (100+M*4,100+I),D(M)
130 PSET STEP(2,0),D(M)
140 NEXT M,I
150 GOTO 150
```

## 4. CHANGEMENT DE JEU DE CARACTERES

Lorsque le module d'extension est connecté au X-07, on peut disposer de trois jeux de caractères.

Pour sélectionner l'un des jeux, sélectionner le mode standard, le mode Hirakana, ou le mode affichage à cristaux liquides (LCD).

Le caractère correspondant à l'un des emplacements de 00 à FF dans le tableau de jeu de caractères est affiché lors de la sortie de CHR\$ (20) suivi du code du caractère.

### Mode standard

Ce mode est automatiquement sélectionné à la mise sous tension.

Tableau des codes de caractères du mode standard.

m \ n	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
0	±	☺		0	@	P	'	p	♠	ì	√	-	夕	≡	ô	ε
1	月	↓	!	1	À	Q	a	q	♥	î	。	ア	チ	△	ó	σ
2	火	↑	"	2	B	R	b	r	♣	í	「	イ	ツ	×	o	θ
3	水	↑	#	3	C	S	c	s	♦	ü	」	ウ	テ	モ	ÿ	K
4	木	↑	\$	4	D	T	d	t	○	ü	、	エ	ト	ヤ	Ç	λ
5	金	+	%	5	E	U	e	u	●	ù	・	オ	ナ	ユ	£	μ
6	土	↑	&	6	F	V	f	v	Ä	û	ヲ	カ	ニ	ヨ	N	P
7	日	-	'	7	G	W	g	w	Å	ú	ア	キ	ヌ	ラ	ñ	π
8	年	↑	(	8	H	X	h	x	ä	é	イ	ク	ネ	リ	Γ	τ
9	円	↑	)	9	I	Y	i	y	à	ë	ウ	ケ	ノ	ル	Σ	φ
A	時	L	*	:	J	Z	j	z	â	è	エ	コ	ハ	レ	Π	χ
B	分	↑	+	:	K	[	k	{	á	ê	オ	サ	ヒ	ロ	Ω	ω
C	秒	→	,	<	L	¥	l	!	ã	é	ト	シ	フ	ワ	α	υ
D	円	←	-	=	M	]	m	}	ä	ö	ユ	ス	ヘ	ン	β	ε
E	円	↑	.	>	N	^	n	~	ï	ö	ヨ	セ	ホ	°	γ	φ
F	円	↓	/	?	O	_	o	÷	ï	ò	ッ	ソ	マ	°	ó	÷



## Mode Hirakana

Le mode Hirakana est sélectionné lorsque l'on a une valeur autre que 0 à l'adresse 4D7H. Les caractères indiqués dans le tableau ci-dessous peuvent alors être affichés. Comme on peut le voir, les caractères correspondant aux emplacements A6 à FA ne sont pas les mêmes qu'en mode standard.

Tableau des codes de caractères du mode Hirakana

n \ m	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
0	±	⊕		⊖	@	P	'	p	♠	ì	√	-	た	み	お	※
1	月	⌞	!	1	A	Q	a	q	♥	î	◦	あ	ち	む	♀	≡
2	火	⌞	"	2	B	R	b	r	♣	í	「	い	つ	め	♂	≡
3	水	⌞	#	3	C	S	c	s	♦	ü	」	う	て	も	∞	—
4	木	⌞	\$	4	D	T	d	t	○	ü	、	え	と	や	※	—
5	金	+	%	5	E	U	e	u	●	ù	・	お	な	ゆ	U	百
6	土		&	6	F	V	f	v	Ä	û	を	か	に	よ	ij	千
7	日	—	'	7	G	W	g	w	Å	ú	あ	き	ぬ	ら	⊗	万
8	年	⌞	(	8	H	X	h	x	ä	é	い	く	ね	り	月	大
9	円	⌞	)	9	I	Y	i	y	à	ë	う	け	の	る	≠	中
A	時	⌞	*	:	J	Z	j	z	â	è	え	こ	は	れ	≧	小
B	分	⌞	+	;	K	[	k	{	á	ê	お	さ	ひ	ろ	《	W
C	秒	→	,	<	L	¥	l	:	å	é	や	し	ふ	わ	》	レ
D	⌞	←	—	=	M	]	m	}	ä	ö	ゆ	す	へ	ん	≈	も
E	⌞	↑	.	>	N	^	n	~	ï	ö	よ	せ	ほ	ゝ	×	♠
F	⌞	↓	/	?	O	_	o	÷	ï	ò	っ	そ	ま	°	■	÷

## Mode affichage à cristaux liquides (LCD)

Le mode LCD est sélectionné lorsque l'on a une valeur autre que 0 à l'adresse 15H. Les caractères indiqués dans le tableau ci-dessous peuvent alors être affichés. En LCD, on peut obtenir l'affichage sur l'écran du téléviseur ou du moniteur des caractères définis par l'utilisateur à l'aide de la fonction FONT\$.

Tableau des codes de caractères du mode LCD

$\begin{matrix} m \\ n \end{matrix}$	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
0	±	⊕		0	@	P	`	F	♣	ì	√	-	☞	≡	ö	ε
1	月	⌞	!	1	A	Q	a	q	♥	ï	•	ア	チ	△	ó	σ
2	火	⌞	"	2	B	R	b	r	♣	í	Γ	イ	ツ	×	◊	θ
3	水	⌞	#	3	C	S	c	s	♣	ü	J	ウ	テ	E	ÿ	κ
4	木	⌞	\$	4	D	T	d	t	○	ü	、	I	ト	Ɔ	φ	λ
5	金	⌞	%	5	E	U	e	u	●	ù	+	オ	ナ	⌒	ς	μ
6	土	⌞	&	6	F	V	f	v	Ä	ü	ヲ	カ	ニ	ヨ	ñ	ρ
7	日	—	'	7	G	W	g	w	Å	ú	フ	キ	ヌ	う	ñ	π
8	年	⌞	(	8	H	X	h	x	ä	é	イ	ク	ネ	リ	Γ	τ
9	円	⌞	)	9	I	Y	i	y	à	ë	ウ	ケ	ノ	ル	Σ	φ
A	時	⌞	*	:	J	Z	j	z	ä	è	エ	コ	ハ	レ	Π	χ
B	分	⌞	+	;	K	[	k	{	á	ë	オ	サ	ヒ	□	Ω	ω
C	秒	→	,	<	L	¥	l	!	á	é	ヤ	シ	フ	ワ	α	ν
D	⌞	←	-	=	M	]	m	}	ä	ö	ユ	ズ	ハ	フ	β	£
E	⌞	↑	.	>	N	^	n	~	ï	ö	ヨ	セ	市	°	Υ	Φ
F	⌞	↓	/	?	O	_	o	÷	ï	ò	ツ	ソ	マ	°	φ	÷

# Annexe

## ■ TABLEAU DES VALEURS LIMITE

Mode		1	2	3
Page		1, 2, 3	1, 2, 3	2
Définition (points)		32×16	64×48	128×192
Taille du point		8×12	4×4	2×2
Nombre de caractères de l'affichage		32 caractères × 16 lignes	32 caractères × 16 lignes	16 caractères × 8 lignes
N° de couleur	0	—	Noir/noir	—
Jeu de couleurs	1	Vert/rouge	Vert/blanc	Vert/blanc
<sup>1</sup> / <sub>2</sub>	2	Vert clair/orange	Jaune/cyan	Jaune/cyan
	3	Rouge/vert	Bleu/violet	Bleu/violet
	4	Orange/vert clair	Rouge/orange	Rouge/orange
	5	—	Blanc/vert	—
	6	—	Bleu/jaune	—
	7	—	Violet/bleu	—
	8	—	Orange/rouge	—
Position d'image	X	0~31	0~31	0~31
	Y	0~15	0~15	0~95

Mode		4	5	6
Page		2	3	3
Définition (points)		128×192	128×192	256×192
Taille du point		2×1	2×1	1×1
Nombre de caractères de l'affichage		16 caractères × 16 lignes	16 caractères × 16 lignes	32 caractères × 16 lignes
N° de couleur	0	Noir/noir	—	Noir/noir
Jeu de couleurs	1	Vert/blanc	Vert/blanc	Vert/blanc
<sup>1</sup> / <sub>2</sub>	2	—	Jaune/cyan	—
	3	—	Bleu/violet	—
	4	—	Rouge/orange	—
	5	—	—	—
	6	—	—	—
	7	—	—	—
	8	—	—	—
Position d'image	X	0~15	0~31	0~31
	Y	0~191	0~191	0~191

## ■ TABLEAU DE SYNTAXE

SCREEN [<mode écran>] [, [<page active>] [, <page visuelle>]]

COLOR [<couleur d'avant-plan>] [, [<couleur d'arrière-plan>] [, <jeu de couleurs>]]

PSET PRESET	[STEP] (<abscisse X>, <ordonnée Y>) [, <N° de couleur>]
----------------	---

LINE [SETP] (<abscisse X>, <ordonnée Y>)-[SETP (<abscisse x>, <ordonnée y>)] [, [<N° de couleur>] [, 

B
BF

]]

CIRCLE [STEP] (<abscisse X>, <ordonnée Y>), <rayon> [, [<N° de couleur>] [, <coefficient de forme>]]

PAINT [STEP] (<abscisse X>, <ordonnée Y>)[, [<N° de couleur>] [, <couleur de pourtour>]]

POINT (<abscisse X>, <ordonnée Y>)

SCREEN (<position d'image X>, <position d'image Y>)

# **Canon** CANON INC.

7-1, Nishi-shinjuku 2-chome, Shinjuku-ku, Tokyo 160, Japan  
P.O. Box 5050, Shinjuku Dai-ichi Seimei Building, Tokyo 160, Japan

## **CANON U.S.A., INC.**

HEAD OFFICE One Canon Plaza, Lake Success, N.Y. 11042, U.S.A.

CHICAGO 140 Industrial Drive, Elmhurst, Illinois 60126, U.S.A.

LOS ANGELES 123 Paularino Avenue East, Costa Mesa, California 92626, U.S.A.

ATLANTA 6380 Peachtree Industrial Blvd., Norcross, Georgia 30071, U.S.A.

DALLAS 2035 Royal Lane, Suite 290, Dallas, Texas 75229, U.S.A.

## **CANON CANADA INC.**

HEAD OFFICE 3245 American Drive, Mississauga, Ontario, L4V 1N4, Canada

CALGARY 2828, 16th Street, N.E. Calgary, Alberta, T2E 7K7, Canada

## **CANON EUROPA N.V.**

P.O. Box 7907, 1008 AC Amsterdam, The Netherlands

## **CANON FRANCE S.A.**

DIVISION CALCUL 93154 Le Blanc Mesnil, Cedex, France

## **CANON RECHNER DEUTSCHLAND GmbH.**

Fraunhoferstrasse 14, Postfach 8033, München-Martinsried, West Germany

## **CANON UK LTD.**

Waddon House, Stafford Road, Croydon CR9 4DD, England

## **CANON LATIN AMERICA, INC.**

SALES DEPARTMENT P.O. Box 7022, Panama 5, Rep. of Panama

REPAIR SERVICE CENTER P.O. Box 2019, Colon Free Zone, Rep. of Panama

## **CANON HONG KONG TRADING CO., LTD.**

Golden Bear Industrial Centre, 7th Floor, 66-82 Chai Wan Kok Street, Tsuen Wan, New Territories, Hong Kong

## **CANON AUSTRALIA PTY. LTD.**

1 Hall Street, Hawthorn East, Victoria 3123, Australia