


FAITES VOS JEUX

AVEC CANON X07



EYROLLES

PHILIPPE IFRAH

*FAITES VOS JEUX
AVEC CANNON X07*

CHEZ LE MÊME ÉDITEUR

Dans la même collection

- | | |
|----------------|--|
| VANRYB-POLITIS | — <i>Tout savoir sur New-Brain</i> , 104 p.; 1984. |
| GOLDSTEIN | — <i>Le guide de l'IBM-PC</i> , 272 p.; 1984. |
| BUNN | — <i>Obtenez le maximum de votre ATARI</i> , 104 p., 1984. |
| GARCIA | — <i>Le Basic minimum</i> . 15 mots pour apprendre à programmer, 104 p., 1984. |
| WANNER | — <i>Aller plus loin en BASIC T07</i> , 312 p., 1984. |
| POLITIS-VANRYB | — <i>Tout savoir sur ORIC</i> , 168 p., 1984. |
| VANRYB-POLITIS | — <i>Tout savoir sur Multitech</i> , 152 p., 1984. |
| DELANNOY | — <i>Faites vos jeux avec ORIC</i> , 224 p., 1984. |
| IFRAH | — <i>Faites vos jeux avec CANON X07</i> , 104 p., 1984. |
| RAVEL | — <i>Tout savoir sur LASER 200-210</i> , 112 p., 1984. |
| VANRYB-POLITIS | — <i>Tout savoir sur LYNX</i> , 176 p., 1984. |
| POLITIS-VANRYB | — <i>Tout savoir sur ATMOS</i> , 192 p., 1984. |

**FAITES VOS JEUX
AVEC CANYON X07**

Si vous désirez être tenu au courant de nos publications, il vous suffit d'adresser votre carte de visite au :

Service « Presse », Éditions EYROLLES
61, Boulevard Saint-Germain,
75240 PARIS CEDEX 05,

en précisant les domaines qui vous intéressent.
Vous recevrez régulièrement un avis de parution
des nouveautés en vente chez votre librairie
habituel.

«La loi du 11 mars 1957 n'autorisant, aux termes des alinéas 2 et 3 de l'article 41, d'une part, que les «copies ou reproductions strictement réservées à l'usage privé du copiste et non destinées à une utilisation collective» et, d'autre part, que les analyses et les courtes citations dans un but d'exemple et d'illustration, «toute représentation ou reproduction intégrale, ou partielle, faite sans le consentement de l'auteur ou de ses ayants droit ou ayants cause, est illicite» (alinéa 1^{er} de l'article 40).»

«Cette représentation ou reproduction, par quelque procédé que ce soit, constituerait une contrefaçon sanctionnée par les articles 425 et suivants du Code pénal.»

Avant-propos


Tous les programmes de cet ouvrage ont été élaborés dans un souci constant de clarté et de qualité, ce qui en rend l'utilisation attrayante.

De nombreux domaines sont abordés : la vie pratique, les mathématiques, l'économie et surtout les jeux; jeux graphiques et musicaux, jeux d'adresse, jeux de réflexion, jeux de société...

L'utilisation de chaque jeu et leur programmation sont largement expliquées, ce qui permet à quiconque de créer ses programmes dans le même ordre d'idées. De plus, un exemple d'utilisation est fourni. A chaque programme vous découvrirez des astuces de programmation mais surtout passerez d'agréables moments.

Je tiens à remercier les nombreuses personnes qui m'ont aidé lors de la réalisation de ce livre, surtout Patrick Zirah et Eric Berton qui m'ont conseillé et ont participé à l'élaboration du présent ouvrage.



UTILISATEUR
DU X07



PROGRAMMEUR
DU X07



CONCEPTEUR
DU X07



X07

Table des matières

Avant-propos	VII
1. Jeux pratiques	1
MENU	1
BIORYTHMES avec prévision des périodes de forme	4
ESPÉRANCE DE VIE selon sexe, âge, métier,etc.	8
GRAPHO	11
2. Jeux graphiques et musicaux	14
Tracé d'une COURBE d'une fonction sur repère réglable	14
COUREUR	17
SYNTHÉTISEUR avec réglages	19
TÉLÉCRAN : dessins à volonté avec effacement point par point	21
CINÉMA MUSICAL: Rock'n Roll avec différents rythmes et mélodies	23

3. Jeux mathématiques	26
FACTORIELLES infinies avec temps de réponse	26
AJUSTEMENT d'une courbe à l'aide de points. 4 jeux en 1	28
- Ajustement par une droite	29
- Par une hyperbole	29
- Par une parabole	30
- Par une exponentielle	30
4. Jeux économiques	35
Votre IMPÔT sur le revenu	35
JEU D'ENTREPRISE, vente d'un produit, possibilités pour plusieurs joueurs ou sociétés, avec un programme pour ceux qui n'ont pas d'imprimante	38
5. Jeux d'adresse	47
FIREFOX avec réglage vitesse et mouvement	47
PACMAN avec poursuite du monstre et apparition de primes	52
6. Jeux de réflexion	58
Le PENDU avec dessin progressif du pendu	58
MÉLI-MÉLO: 7 jeux en 1. Différents types de casse-têtes	61
L'ignoble SCHNOEKLE cache-cache de stratégie	67
7. Jeux de société	73
Jeu de DAMES 3 niveaux	73
BACKGAMMON	80
8. En cas d'erreur	91

1

Jeux pratiques

Menu

Il est indispensable d'établir un menu afin de disposer de plusieurs jeux et sous-programmes à l'intérieur d'un même programme dans votre CANON X07.

Exemple :

- | | |
|---------------------------------------|----------------------------------|
| 1) Grapho
2) Firefox
3) Coureur | 4) Piano
5) Trace
6) Autre |
|---------------------------------------|----------------------------------|

Il suffira ensuite de presser l'une des touches 1, 2, 3, 4, 5 ou 6 pour que le jeu correspondant apparaisse.

Programme :

```
5 FONT$(233)="4,4,4,4,4,4,4,4"
10 PRINT" *** MENU ***"
15 CONSOLE,,,1,1
20 PRINT"1)Grapho | 4) Piano"
30 PRINT"2)Firefox | 5) Trace"
40 PRINT"3)Couleur | 6) Autre"
50 C$=INKEY$
60 IF C$="" THEN 50
70 Y=VAL(C$)
80 ON Y GOTO 100,2000,500,1000,4000,9000
,7000,8000,5000
90 GOTO 10
100 PRINT" GRAPHO"
2000 PRINT" FIREFOX"
500 PRINT" ** COUREUR **"
1000 PRINT" PIANO "
9000 PRINT" *** MENU SUITE ***"
```

Explications :

- 5 La fonction FONT\$ sera expliquée ultérieurement dans le jeu "Grapho".
- 15 Remise en place du son de déclic émis par les touches. Voir "Piano".
- 20 Pour obtenir le trait de séparation, pressez les touches **GRAPH** puis **1**. Cela correspond au caractère 233 défini en 5. Ne vous inquiétez pas s'il apparaît un autre caractère. Le trait vertical apparaîtra lors de l'exécution du programme.
- 50 INKEY\$ affectera à C\$ la valeur de la touche pressée sous forme de caractère.
- 60 Le programme attend qu'une touche soit pressée.
- 70 On affecte la valeur du caractère pressé à Y.
- 80 Lorsque Y prendra la valeur 1 on se rendra à la ligne 200.
Lorsque Y prendra la valeur 2 on se rendra à la ligne 2000.
Et ainsi de suite. Il n'est pas nécessaire que les adresses indiquées

soient dans l'ordre numérique croissant. Cependant, les valeurs de Y doivent être, elles, dans l'ordre 1, 2, 3, 4...

90 Lorsque Y est différent de 1, 2, 3, 4, 5 ou 6 le menu réapparaît.



Biorythmes


Combien d'entre nous ont des périodes de très bonne forme ou des périodes de dépression ?

Il est possible de prévoir plus ou moins ces périodes, car elles sont dues pour 40 % aux événements extérieurs: climat, bonne ou mauvaise nouvelle, chance... Mais il reste 60 % qui étaient jusqu'à présent imprévisible car liés à un rythme intérieur.

Des chercheurs, à partir de plusieurs expériences, ont découvert que ces hauts et bas de votre forme étaient périodiques; ils ont donc cherché à en établir les périodes. C'est ainsi que naît cette nouvelle science: les biorythmes.

D'après les expériences, la période de forme physique est de 23 jours, celle de la sensibilité 28 jours et 33 jours pour la forme intellectuelle (ou cérébrale). Cf. figure ci-dessous.

Note sur 20



Programme :

```
1000 PRINT" ** BIORYTHMES **"
1020 PRINT"Date de naissance :"
1030 INPUT"Jour";J
1040 INPUT"Mois";M
1050 INPUT"Année";A
1060 S=(A-1901)*365.25+(M-1)*30.44+J-1
1070 IF K=1 THEN 1110
1080 T=S:K=1
1090 PRINT"Date voulue :"
1100 GOTO 1030
```

```
1110 U=S-T
1120 Z=23
1130 GOSUB 1300
1140 PRINT "Physique : ";N;" /20"
1150 Z=28
1160 GOSUB 1300
1170 PRINT "Sensibilité : ";N;" /20"
1180 Z=33
1190 GOSUB 1300
1200 PRINT "Cérébral : ";N;" /20"
1205 IF INKEY$="" THEN 1205
1210 INPUT "1)Changement de date 2)Prochain  
date de bonne forme";C
1220 IF C=1 THEN 1030
1230 IF C<>2 THEN 1210
1240 GOTO 1400
1300 N=SIN((U/Z)-INT(U/Z))*2*3.14159)
1310 N=INT(100*N)/100
1320 N=INT(N*10)+10
1330 RETURN
1400 U=U+1
1410 Z=23
1420 GOSUB 1300
1430 P=N
1435 Z=28
1440 GOSUB 1300
1450 S=N
1455 Z=33
1460 GOSUB 1300
1465 PRINT P;S;N
1470 IF 2*P+S+3*N<90 THEN 1400
1475 U=U+T
1480 AN=INT(U/365,25)
1490 MO=INT(INT(U-AN*365,25)/30,44)
1500 JO=INT(U-AN*365,25-MO*30,44)
1510 PRINT JO+1;" /";MO+1;" /";AN+1901
1520 GOTO 1210
```

Explications :

1060 S : Nombre de jours entre le 1^{er} janvier 1901 et la date J/M/A.

1110 V = Nombre de jours de vie.

1120 Z : période.

1300 N : fonction sinusoïdale de Z.

1320 N : note sur 20.

1470 Pour la date de bonne forme, on donne un coefficient d'importance de la forme :

physique de 2

sensible de 1

cérébral de 3

Avec une moyenne de 15 on est en bonne forme.

JE SUIS JO JO "LA CLASSE"
ET J'EN PROFITE CAR MON
BIORYTHME EST AU MEILLEUR
DE SA FORME INTELLECTUELLE
ET PHYSIQUE

EH BEN !!
QU'EST-CE QUE SA DOIT
ÊTRE D'HABITUDE !



Espérance de vie

Savez-vous combien d'années en moyenne selon : l'âge, le sexe, le métier... vous reste-il à vivre ?

Ce serait une donnée intéressante pour vos projets d'avenir !

Ce programme se base sur des données statistiques concernant l'espérance de vie à partir desquelles des fonctions ont été établies.

Ainsi il est possible de connaître combien d'années il nous reste à vivre.

Au cours de l'exécution du programme, lorsque vous ne connaîtrez pas la réponse à certaines questions (par exemple au sujet de votre métier si vous êtes étudiant) vous aurez la possibilité de donner une réponse intermédiaire ou prévisible.

Programme:

```
100 PRINT" ESPERANCE DE VIE"
200 INPUT"Votre age";A
250 IF A<0 THEN 200
260 IF A>159 THEN 200
270 IF A<1 THEN A=1
300 INPUT"Votre sexe:1)Masculin, 2)Feminin";S
400 INPUT"Votre travail:1)Tres manuel, 2)Manuel, 3)Peu manuel";T
500 INPUT"Vous travaillez dans le secteur:1)Public, 2)Prive, 3)ne sait pas";P
510 IF P=1 THEN Q=1
520 IF P=2 THEN Q=0
530 IF P=3 THEN Q=.5
600 E=-.922*A+69.89
700 IF A<50 THEN 1000
800 E=-.596*A+53.58
850 IF A<80 THEN 1000
900 E=-.25*A+25.94
```

```
930 IF A<99 THEN 1000
950 E=-0.0187*A+3
1000 F=3*T-5+5.3*(S-1)+Q
1200 IF A>40 THEN F=E*F/33
1300 E=E+F
1400 E=INT(E*100)/100
1500 PRINT"Votre esperance de vie:";E;"ans"
1600 IF INKEY$<>"" THEN 100
1700 GOTO 1600
```

Explications :

600 à 900 Courbes et droites ajustées à l'aide des statistiques de l'INSEE et du programme ajustements.

Exemple :

Vous êtes un homme de 40 ans au chômage.

Votre âge ? 40.


Votre sexe 1) Masculin 2) Féminin ? 1

Votre travail: 1) peu manuel 2) manuel 3) très manuel ? 2

Secteur: 1) Privé 2) Public 3) Ne sait pas ? 3

Espérance de vie: 34,51 ans soit 34 ans et demi.

TOUTES LES ESPÉRANCES DE VIE
CALCULÉES , JE N'HÉRITERAIS DE
TANTE MAGGIE QUE DANS 30 ANS




GRAPHO

Objectifs :

A la fois divertissant et utile, ce jeu permet de définir le caractère de votre choix de façon instantanée.

Exemple :

Définissons *Le courreur* sur 6×8 cases :



				0
	1	0	1	0
	1	1	1	0
1	0	1	0	0
		1	0	0
1	1	0	1	0
		1	0	0

Préliminaires :

Dans le tableau page 93 sont indiqués les caractères définis initialement. Pour calculer le numéro d'appel N correspondant à un caractère donné de ce tableau, on procède ainsi : soient x et y les abscisses et ordonnées du caractère choisi :

$$N = (16.x) + y$$

Il suffira alors de presser :

?CHR\$(N) RETURN

pour obtenir le caractère choisi.

Exemple :

Pour appeler le caractère "é":
sur le tableau: abscisse = 7;
ordonnée = 15

d'où :

$$N = (16 \times 2) + 15 = 127$$

? CHR\$(127) **RETURN**

Si l'on définit le coureur au numéro 233, il suffit de presser les touches **GRPH** puis **1** pour l'obtenir. Ne sont redéfinissables que les caractères compris entre 128 et 159 ou entre 224 et 271.

Programme:

```
100 PRINT" GRAPHO"
110 DIM A(8)
120 FOR I=1 TO 8
130 PRINT"Line numero ";I;" en base 2";
140 INPUT A
160 A=100*A
170 B=A:D=0:E=0
180 B=INT(B)
190 B=B/10
200 C=10*(B-INT(B))
210 IF C>1 THEN 350
220 D=D+C*2^E
230 E=E+1
240 IF E<8 THEN 180
260 A(I)=D
265 PRINTD;"en base 10"
270 NEXT I
280 PRINT"Caractere numero";
290 INPUT G
300 ON ERROR GOTO 280
310 FONT$(G)="A(1),A(2),A(3),A(4),A(5),A
(6),A(7),A(8)"
320 PRINTCHR$(G)
330 END
350 PRINT"ERREUR " :GOTO 130
```

Explications :

- 110 On définit 8 mémoires numérotées A(1), A(2), ..., A(8).
- 150 Il ne peut y avoir plus de 6 points par ligne du caractère.
- 260 A(I) prendra successivement les valeurs de Ax100 en base 10.
- 280 Si l'on veut que le caractère reste défini, on choisira un numéro de caractère définissable (obtenus par **GRPH**).
- 310 FONT\$ permet de définir le caractère. En prenant l'exemple du coureur, on aura:
FONT\$(233) = "0, 40, 112, 160, 32, 208, 16, 0".

En faisant: ? CHR\$(233), on obtient le dessin du coureur.

- 300 Lorsqu'un code n'est pas définissable par l'utilisateur, la machine devrait mentionner "erreur" à la ligne 310. Mais ici on préférera en cas d'erreur que la question soit reposée.

2

Jeux graphiques et musicaux

Tracé d'une courbe

Dans un repère orthogonal, la courbe de votre choix se dessine sous vos yeux.

Soit f une fonction polynôme réelle donnée du type :

$$f(X) = AX^4 + BX^3 + CX^2 + DX + E$$

Vous choisissez les valeurs A, B, C, D, E ainsi que les vecteurs unitaires des axes Ox et Oy . Sur votre écran se dessine la courbe de la fonction f dans le repère choisi.

Programme :

```
8000 PRINT "TRACE D'UNE COURBE"
8010 PRINT "Courbe : Y=AX^4+BX^3+CX^2+DX+E
"
```

```
8020 INPUT"A=" ;A
8030 INPUT"B=" ;B
8040 INPUT"C=" ;C
8050 INPUT"D=" ;D
8060 INPUT"E=" ;E
8061 INPUT"Echelle Y:Nombre de points/une
ite";U
8062 INPUT"Echelle X:Nombre de points/une
ite";V
8063 IF U*U=0 THEN 8061
8064 CLS
8066 U=1/U
8067 V=1/V
8068 LINE(0,24)-(120,24)
8069 LINE(40,0)-(40,32)
8070 FOR X=0 TO 119
8075 Z=(X-40)/V
8080 Y=A*Z^4+B*Z^3+C*Z^2+D*Z+E
8085 Y=Y/U
8090 Y=23-INT(Y)
8110 IF Y>31 THEN 8160
8120 IF Y<0 THEN 8160
8125 IF X=0 THEN 8150
8130 IF ABS(S-Y)<2 THEN 8150
8135 ON ERROR GOTO 8150
8137 IF ABS(S-Y)>9 THEN S=Y
8140 LINE(X-1,S)-(X,Y)
8150 PSET(X,Y)
8155 S=Y
8160 NEXT X
8200 IF INKEY$="" THEN 8200
8250 INPUT"Changement d'echelle (O/N)";C
$
8260 IF C$="O" THEN 8061
8270 IF C$="N" THEN 8000
8280 GOTO 8250
8300 8000
```

Explications :

- 8061 U sera le nombre de points de l'écran contenus dans une unité de l'axe des Y, donc dans une unité verticale.
- 8062 V sera le nombre de points de l'écran contenus dans une unité de l'axe des X, soit dans une unité horizontale.
- 8064 Fonction d'effacement de l'écran.
- 8068 Tracé de l'axe des X.
- 8069 Tracé de l'axe des Y.
- 8070 On parcourt l'ensemble des points de l'axe des X.
- 8110 à 8120 On évite de sortir de l'écran.
- 8130 à 8140 On trace une ligne verticale entre deux points successifs trop éloignés.
- 8150 Le point de coordonnées X, Y est imprimé.

Exemple :

Prenez les valeurs :

A = -1

B = 1

C = 8

D = -9

E = 1

Échelle Y: -0,2

Échelle X: 0,2

Et observez la courbe qui se dessine sur votre écran.

A vous !

Le coureur en action

Le coureur est un exemple de dessin animé, simple, qui vous permettra d'en réaliser de plus sophistiqués en vous basant sur ce modèle.

Programme :


```
500 PRINT" ** COURREUR **"
501 FONT$(153)="0,40,112,160,32,208,16,0
"
502 FONT$(136)="0,4,4,4,4,12,4,0"
505 PRINT"";
510 J=J+1
515 IF J>19 THEN J=0
520 FOR I=1 TO 100:NEXT I
530 PRINTCHR$(&H0C)
540 FOR I=1 TO J-1:PRINT" ";
545 NEXT I
550 PRINT"";
560 FOR I=1 TO J:PRINT" ";
```

Pour les caractères graphiques consultez les explications.

```
570 PRINTCHR$(&H0C)
580 FOR I=1 TO J:PRINT" ";
590 NEXT I
600 GOTO 505
```

Explications :

- 501 On affecte le dessin du coureur au caractère 153 correspondant aux touches **GRPH** puis **E**. Voir GRAPHO.
- 502 On affecte le dessin suivant au caractère 136 correspondant aux touches **GRPH** puis **A**.



*Coordonnées
figure*

0
1
1
1
1
11
1
0

*Coordonnées
grapho*

0
4
4
4
4
12
4
0

- 505 On presse les touches **GRPH** puis **E** pour le caractère entre guillemets.
- 510 J: Nombre d'espaces parcourus par le coureur.
- 550 Pressez les touches **GRPH** puis **A**.
- 570 Cette fonction efface l'écran.

Synthétiseur


Un véritable synthétiseur à votre disposition grâce à un programme relativement court.

Les touches sont disposées comme sur un vrai piano (cf. figure). Les bémols y sont représentés par des cases sombres.

La touche **1** raccourcit la durée du son.

La touche **]** allonge la durée du son.

La touche d'espacement arrête le synthétiseur.



Programme:

```
10 CLS
50 PRINT " ** SYNTETISEUR **"
100 CONSOLE,,,0
150 C=5
200 A$=INKEY$
210 IF A$="" THEN 200
220 IF A$="1" THEN C=C-1
230 IF A$"]=" THEN C=C+1
```

```
240 IF A$="" THEN 700
250 IF C<1 THEN C=1
400 R=INSTR("Q2W3ER5T6Y7UI900P@^AZSXCFUGBNJMK,L./?:?",A$)
500 BEEP R,C
550 PRINT A$;
600 GOTO 200
700 CONSOLE,,,1,1
```

Explications:

- 100 On retire le cliquetis des touches.
 - 150 On initialise la durée du son.
 - 400 A prend les valeurs 1, 2, 3, 4, ... lorsque A\$ prend respectivement les valeurs Q, 2, W, 3, ...
 - 500 Son.
 - 550 On fait apparaître sur l'écran la touche pressée.
 - 700 Remise en place du cliquetis des touches.

Utilisation :

Presser **1** deux fois pour raccourcir la durée du son. Puis jouez "Au clair de la lune":

**I I I O P O I P O O I
O O O O Y Y O I U Y T P** (deux fois)
(et recommencez)

Télécran : Dessins à volonté

Objectif :

Vous allez pouvoir grâce à ce programme réaliser les dessins de votre choix sur la totalité de votre écran, soit 120×32 points.

Il suffit de promener le curseur à droite, à gauche, vers le haut ou vers le bas. L'effacement est possible point par point et de façon instantanée.

Programme :

```
4000 CLS
4020 C$=INKEY$
4030 IF C$="]" THEN A=A+1
4040 IF C$=":" THEN A=A-1
4050 IF C$="[" THEN B=B-1
4060 IF C$="?" THEN B=B+1
4070 IF C$="0" THEN D=0
4080 IF C$="1" THEN D=1
4090 IF A<0 THEN A=0
4100 IF B<0 THEN B=0
4110 IF B>31 THEN B=31
4120 IF A>119 THEN A=119
4130 IF D=1 THEN PSET(A,B)
4140 IF D=0 THEN PRESET(A,B)
4150 GOTO 4020
```

Explications :

4000 Cette fonction efface l'écran.

4090 à 4120 On évite de sortir de l'écran.

4130 Si la touche a été pressée, le point de coordonnées A,B s'imprime.

4140 La fonction PRESET efface le point de coordonnées A,B.

Utilisation :

RUN

Pour dessiner: Presser la touche **1** (Un point apparaît sur le coin supérieur gauche).


Pour effacer: Presser la touche **0**

Pour aller: à droite presser **]**

à gauche presser **:**

Pour aller: vers le haut presser **[**

vers le bas presser **?**



Cinéma musical: Rock'n Roll

Présentation :

Sur votre écran, deux couples dansent le Rock'n Roll sur une musique connue. Impressionnant !!

Un petit programme truffé d'astuces permet ce résultat. Vous pourrez utiliser par la suite ces astuces de programmation qui vous ouvriront de nombreux horizons.

Ébahis, vous constaterez qu'en pressant la touche **[2]** le rythme musical s'accélère ; *les danseurs suivent !* Pressez **[2]** plusieurs fois ; le rythme s'accélère davantage. Pour ralentir le rythme pressez **[1]**.

Encore plus fort ! Vous pouvez changer d'octave en pressant la touche **[^]**. Enfin, si vous pressez la touche **[-]** : *Surprise !! Changement de mélodie !*

A vous de jouer avec ces quatre touches pour obtenir les mélodies désirées pendant que les deux couples infatigables de danseurs suivent votre rythme.

Programme :

```
100 PRINT" * CINEMA MUSICAL *"
110 FOR I=0 TO 500:NEXT I
130 CONSOLE,,,0
140 FONT$(224)="0,16,28,16,16,16,40,72"
150 FONT$(225)="0,40,240,32,32,32,88,64"
160 FONT$(228)="0,84,56,16,16,16,104,8"
170 FONT$(229)="0,160,112,40,32,32,80,72
"
180 FONT$(231)="4,8,4,4,4,4,8,8"
190 FONT$(233)="128,104,48,160,32,32,144
,80"
195 C=5:A=0:D=1:E=1
200 Z$=INKEY$
205 IF Z$="^" THEN D=-D
210 IF Z$="1" THEN C=C+1
```

```

215 IF Z$="-" THEN E=-E
220 IF Z$="2" THEN C=C-1
225 IF C<1 THEN C=1
230 B=B+1
240 IF B=3 THEN A=5
250 IF B=5 THEN A=0
260 IF B=7 THEN A=7
270 IF B=8 THEN A=5
280 IF B=9 THEN B=1 :A=0
290 CLS
295 IF D<0 THEN A=A+12
300 LOCATE 8,1
320 PRINT" xx xx ";
330 IF E>0 THEN BEEP 1+A,C+2:BEEP A+5,C+
2:BEEP A+8,C+2
340 IF E<0 THEN BEEP 1+A,C:BEEP A+5,C+4:
BEEP A+8,C:BEEP 1+A,C+3
350 CLS
360 LOCATE 8,1
380 PRINT" xx xx ";
390 IF E<0 THEN BEEP A+5,C+1:BEEP A+8,C+
4
400 IF E>0 THEN BEEP A+10,C+2:BEEP A+11,
C+2
410 CLS
420 LOCATE 8,1
440 PRINT" xx xx ";
450 IF E>0 THEN BEEP A+10,C+2: BEEP A+8,
C+2
460 IF E<0 THEN BEEP 1+A,C:BEEP A+5,C+1:
BEEP 0,2:BEEP A+8,C+5
500 CLS
510 LOCATE 8,1
530 PRINT" xx xx ";
540 IF E>0 THEN BEEP A+5,C+3
550 IF E<0 THEN BEEP 1+A,C:BEEP A+5,C+1:
BEEP 0,2:BEEP A+8,C+8
600 GOTO 200

```

Pour mettre en place
les caractères
graphiques reportez-
vous aux explications
aux numéros de lignes
correspondantes.

Explications :

110 Légère attente.

130 On annule le cliquetis des touches.

140	On crée le caractère		qui s'obtient en faisant	GRPH	7
150	On crée le caractère		qui s'obtient en faisant	GRPH	Z
160	On crée le caractère		qui s'obtient en faisant	GRPH	C
170	On crée le caractère		qui s'obtient en faisant	GRPH	J
180	On crée le caractère		qui s'obtient en faisant	GRPH	@
190	On crée le caractère		qui s'obtient en faisant	GRPH	1

195 On initialise à 5 la durée du son.

230 B: Niveau de la mélodie.

300 Localisation du curseur.

320 Pour obtenir le dessin entre les guillemets, on presse les touches suivantes :

étant la touche d'espacement.

Ne vous inquiétez pas si les dessins n'apparaissent pas tout de suite ; ils apparaîtront lorsque le programme tournera.

380	Touches:								
440	Touches:								
530	Touches:								

3

Jeux mathématiques

Factorielles infinies

Définition d'une factorielle :

$$0! = 1$$

$$N! = 1 \times 2 \times 3 \times 4 \dots \times (N-1) \times N$$

Objectif :

Peu nombreuses sont les machines qui calculent les factorielles. Mais en connaissez-vous qui calculent ($500!$) ?

Au-delà de $69!$, la capacité des calculatrices de poche est dépassée ($69! = 1,7112245 \times 10^{98}$).

Le Canon X07 n'a pas vocation de calculateur scientifique, et pourtant un programme assez simple vous permettra de calculer des factorielles impressionnantes !

Le temps de calcul est parfois assez long pour les factorielles de nombres élevés, le Canon X07 a la courtoisie de vous annoncer à l'avance le temps qu'il lui faudra pour effectuer ce calcul à la seconde près !!!

Programme :

```
7000 PRINT"FACTORIELLES INFINIES"
7010 INPUT"Factorielle de ";N
7020 IF N<0 OR N>INT(N) THEN 7010
7030 IF N=0 THEN PRINT"0!= 1":GOTO 7010
7035 IF N>100 THEN 7200
7040 Q=N:W=0
7050 FOR I=1 TO N-1
7060 Q=Q*I
7080 IF Q>=10 THEN Q=Q/10:W=W+1:GOTO 708
0
7100 NEXT I
7120 PRINTN;"! =" ;Q ;"*10^" ;W ;
7130 INPUT"      " ;P
7140 GOTO 7000
7200 X=INT(.06*N^1.155)
7210 R=INT(X/60)
7220 T=X-R*60
7230 PRINT"Attente :" ;R ;"mn." ;T ;"sec";
7240 PRINT" - Voulez-vous toujours ";N;
7250 INPUT"! (O/N)" ;U$
7270 IF U$="O" THEN 7040
7280 IF U$="N" THEN 7000
7300 GOTO 7230
```

Explications :

7020 On ne considère que les nombres entiers positifs.

7030 Par définition $0!=1$.

- 7035 Le temps d'attente ne sera indiqué que lorsque l'utilisateur demandera le calcul d'une factorielle supérieure à 100.
- 7040 Q est la partie réelle du résultat.
W est la partie en exposant du résultat.
 $N! = Q \times 10^W$.
- 7080 Lorsque Q est supérieur ou égal à 10 on divise Q par 10 et l'exposant s'accroît d'une unité. Ainsi la valeur finale affectée à Q est toujours comprise entre 0 et 10.
- 7200 Le nombre X de secondes d'attente est fonction du N choisi. La fonction utilisée est : $X = 0,06 \times N^{1,155}$.
- 7210 à 7220 Séparation Minutes/Secondes.

Exemple :

Factorielle de ? 70
 $70! = 1.197857166997 \times 10^{100}$

[Return](#)

Factorielle de ? 170
 Attente : 0 mn 22 sec —
 Voulez-vous toujours 170 ! (O/N) ? O
 $170! = 7.2574156153103 \times 10^{306}$

[Return](#)

[Return](#)


[Return](#)

Ajustement d'une courbe à l'aide de points


A partir d'un nuage de points il est souvent très utile d'ajuster la droite ou la courbe qui représente le mieux ce nuage. Cela donne une forme générale à une série statistique et permet de faire des prévisions ou de se renseigner sur un point donné sans disposer de l'ensemble des points.

Ainsi, ce programme permet d'ajuster au mieux la courbe désirée, droite, hyperbole, parabole ou exponentielle, et de faire des prévisions qui suivraient la même tendance.


*Ajustement par
une droite*




*Ajustement par
une hyperbole*



*Ajustement par
une parabole*



*Ajustement par
une exponentielle*



Programme:

```
10 PRINT "AJUSTEMENT D'UNE COURBE"
20 INPUT "Nombre de points"; N
30 FOR I=1 TO N
40 INPUT "Abscisse"; A
50 INPUT "Ordonnee"; O
55 Z=LOG(10)
60 B=B+A
70 C=C+O
80 D=D+A^2
90 E=E+O^2
100 F=F+A*O
120 H=H+O*A^2
130 K=K+A^3
150 M=M+A^4
160 IF O<=0 THEN 200
165 XA=XA+LOG(A)/Z
170 U=U+LOG(O)/Z
175 XC=XC+(LOG(A)/Z)^2
180 S=S+A*LOG(O)/Z
190 XY=XY+LOG(A)*LOG(O)/Z^2
200 NEXT I
210 PRINT "Type de courbe:"
220 PRINT "- Droite (1), Courbe hyperbolique (2),"
225 INPUT " Parabolique (3) ou exponentielle (4)"; P
230 ON P GOTO 235, 400, 600, 800
232 GOTO 210
235 IF D-B^2=0 THEN 622
240 U=(F-B*C/N)/(D-B*B/N)
250 W=C/N-U*B/N
260 PRINT "Y="; U; "*X+"; W
265 IF INKEY$="" THEN 265
270 PRINT "Prévision (1), changement de courbe (2),"
280 INPUT " ou fin (3)"; X
```

```

290 ON X GOTO 310, 210, 1000
300 GOTO 270
310 ON P GOTO 315, 500, 750, 900
315 INPUT "Abscisse :X=" ;A
320 O=A*U+W
330 PRINT "Ordonnee :Y=" ;O
340 IF INKEY$="" THEN 340
350 GOTO 270
400 IF D-B^2=0 THEN 622
405 XM=XA/N :YM=U/N
410 U=(XY-XM*U)/(XC-XM*X)
420 W=U/N-V*X/A/N
430 W=10^W
450 PRINT "Y=" ;W ;" *X^" ;U
460 IF INKEY$="" THEN 460
470 GOTO 270
500 INPUT "Abscisse :X=" ;X
510 Y=W*X^U
520 PRINT "Ordonnee Y=" ;Y
525 IF INKEY$="" THEN 525
530 GOTO 270
600 U=M*(D*N-B^2)-K*(K*N-D*B)+D*(K*B-D^2)
)
620 IF U<>0 THEN 630
622 PRINT "Pas de solution ;"
624 IF INKEY$="" THEN 624
626 GOTO 210
630 W=K*(B*C-F*N)-D*(D*C-F*B)+H*(D*N-B^2)
)
640 W=W/U
650 X=D*(F*D-K*C)-H*(B*D-K*N)+M*(B*C-F*N)
)
660 X=-X/U
670 Y=H*(K*B-D^2)-M*(F*B-D*C)+K*(F*D-K*C)
)
680 Y=Y/U
690 PRINT "Y="
700 PRINT W;" *X^2+" ;X;" *X+" ;Y

```

```

710 IF INKEY$="" THEN 710
720 GOTO 270
750 INPUT "Abscisse X=";L
760 L=W*L^2+X*L+Y
770 PRINT "Ordonnée Y=";L
775 IF INKEY$="" THEN 775
780 GOTO 270
800 EX=(S-B*L/N)/(D-B*B/N)
810 EB=(L-EX*B)/N
820 AX=10^EX
830 BX=10^EB
840 PRINT "Y=";BX;"*";AX;"^X"
850 IF INKEY$="" THEN 850
860 GOTO 270
900 INPUT "Abscisse X=";A
910 Y=B*X*A*X^A
920 PRINT "Ordonnée Y=";Y
950 IF INKEY$="" THEN 950
980 GOTO 270
1000 END

```

Explications :

- 10 à 200 Définition des variables.
- 210 à 230 Choix de la courbe.
- 270 à 310 Choix, prévision ou changement de courbe.
- 235 à 350 Droite.
- 400 à 530 Courbe hyperbolique.
- 600 à 780 Courbe parabolique.
- 800 à 980 Courbe exponentielle.

Exemple :

Par exemple supposons la série statistique de la dette y en milliards de dollars constants des pays en voie de développement (Source OCDE) :

<i>Année</i>	<i>x</i>	<i>y</i>
1971	1	116
1975	5	180
1976	6	205
1977	7	235

En dessinant un rapide graphique on s'aperçoit qu'il s'agit d'une fonction exponentielle.

Ainsi, en utilisant le programme :

Nombre de points ?	4
Abscisse ?	1
Ordonnée ?	116
Abscisse ?	5
Ordonnée ?	180

Et ainsi de suite ; Puis vous choisisrez le type de courbe :

Dans ce cas pour "exponentielle" vous presserez : **4**

Et on obtient la courbe $y = 102,57(1,123)^x$.

Pressez une touche quelconque, puis vous avez le choix entre faire une prévision et un changement de courbe.

Si l'on fait une prévision : $x = 9$ (soit 1979) on obtient $y = 291,76$. (Ce qui est près de la réalité : 294).

Ajusté par une droite on aurait obtenu pour 1979 $y = 264,89$.

Ce qui correspond peu à la réalité.

Ajusté par une parabole on aurait obtenu pour 1979 : $y = 275,36$.

Par une hyperbole : $y = 236,89$.

4

Jeux économiques

Calcul de votre impôt

Voici un moyen rapide de calculer ses impôts et de voir combien payent les autres, selon leurs gains et leur situation familiale.

Bien entendu, ce programme ne tient pas compte des cas très particuliers d'allègements fiscaux ou autres.

Programme:

```
100 PRINT"* CALCUL DE VOTRE * IMPOT SUR  
LE REVENU"  
150 IF INKEY$="" THEN 150  
200 INPUT"Nombre de personnes dans votre  
foyer fiscal";P
```

```

300 IF P<=0 OR P>INT(P) THEN 200
400 IF P>2 THEN P=P/2+1
450 PRINT"Cela fait";P;"parts"
500 INPUT"Quel est votre revenu annuel n
et global en Francs";R
600 R=10*INT(R/10)
650 Q=R/P
700 IF Q<=12620 THEN I=0:GOTO 3000
800 IF Q<=13190 THEN I=(R*.05)-(631*P):G
OTO 3000
900 IF Q<=15640 THEN I=(R*.1)-(1290.5*P)
:GOTO 3000
1000 IF Q<=24740 THEN I=(R*.15)-(2072.5*
P):GOTO 3000
1100 IF Q<=31810 THEN I=(R*.2)-(3309.5*P
):GOTO 3000
1200 IF Q<=39970 THEN I=(R*.25)-(4900*P)
:GOTO 3000
1300 IF Q<=48360 THEN I=(R*.3)-(6898.5*P
):GOTO 3000
1400 IF Q<=55790 THEN I=(R*.35)-(9316.5*
P):GOTO 3000
1500 IF Q<=92970 THEN I=(R*.4)-(12106*P)
:GOTO 3000
1600 IF Q<=127860 THEN I=(R*.45)-(16754.
5*P):GOTO 3000
1700 IF Q<=151250 THEN I=(R*.5)-(23147.5
*P):GOTO 3000
1800 IF Q<=172040 THEN I=(R*.55)-(30710*
P):GOTO 3000
1900 IF Q<=195000 THEN I=(R*.6)-(39312*P
):GOTO 3000
2000 I=(R*.65)-(49062*P)
3000 PRINT"Votre impot a payer :";I;"Fran
cs"
4000 IF INKEY$="" THEN 4000
4100 GOTO 100

```

Explications :

- 400 Au-delà de la deuxième personne, chaque personne à charge donne une demi-part.
- 700 Lorsque le quotient familial annuel est inférieur à 12 620 F, on ne paie pas d'impôt.

Exemple :

Vous êtes marié, père de trois enfants et votre revenu annuel net global est de 130 000 F.

RUN puis pressez une touche quelconque.

Nombre de personnes dans votre foyer fiscal ? 5

Cela fait 3,5 parts.

Revenu annuel net global ? 130 000.

Votre impôt à payer: 15 350 F.



Jeu d'entreprises

Objectif :

Les joueurs mettent à l'épreuve leur talent de chefs d'entreprise dans un marché où la concurrence est déterminante. (1 à 6 joueurs).

Règles du jeu :

Chaque joueur dispose de deux sociétés. Chaque société démarre avec un capital initial de 60 000 F.

Le but du jeu est de dégager des profits dans la vente d'un guide touristique.

Une société peut avoir de graves problèmes si elle dose mal ses investissements. Il est possible d'emprunter, mais l'emprunt coûte cher (20 % d'intérêt par an).

Lorsque le capital restant ne couvre plus les frais fixes (main d'œuvre, amortissements des machines, loyer des locaux, ...) la société fait faillite.

Déroulement de la partie :

A) Entrée des données

- Indiquer le nombre de sociétés et leurs dénominations.
- Lorsque le nom d'une société apparaît, seul le propriétaire de cette société est autorisé à voir l'écran de l'ordinateur.
- Après pression sur une touche quelconque de l'appareil, il doit répondre aux questions qui lui sont posées :
- Nombre d'articles fabriqués ? (frais de fabrication : 10 F par article.) ; *Exemple* : 3000.
- Supplément de qualité ? (Supplément ajouté aux 10 F de coût pour une meilleure qualité de produit) ; *Exemple* : + 2.
- Nombre de points de vente ? (Les frais de gestion d'un point de vente sont de 100 F par an, mais il est indispensable de choisir suffisamment de points de vente pour écouter la production) ; *Exemple* : 80.

ALORS MAINTENANT J'ACHÈTE
MOULINEX, GILLETTE ET MANPOWER
ET J'INONDE LE MARCHÉ !!



- Le budget marketing-publicité a des effets sur la période en question, mais une partie de ces dépenses agira encore les années suivantes; *Exemple*: 6 000 F.
- Prix de vente au public: Ce prix doit au moins couvrir les dépenses et dégager un profit, mais s'il est trop élevé par rapport aux concurrents, il sera très difficile de vendre; *Exemple*: 30 F.
- Le montant des intérêts du découvert apparaît sur l'écran. Pour continuer presser une touche quelconque. Le nom d'une nouvelle société apparaît.

B) Résultats par société

Tous les joueurs ayant entré leurs données pour l'année en question, les résultats apparaissent. Cependant ils sont confidentiels. Seul le propriétaire regarde les résultats de sa société.

Après les résultats, il est possible de faire une pause momentanée ou même pour quelques jours (répondre O). Pour éteindre la machine faire:

SLEEP **[Return]**

Si l'on veut reprendre la partie faire:

ON CONT **[Return]**

C) Si vous possédez une imprimante

Vos résultats annuels s'imprimeront automatiquement. Prenez donc soin de conserver l'imprimante sous tension.

Les résultats imprimés sont à la disposition de tous les concurrents:

- Prix de vente,
- Budget approximatif Marketing-Publicité,
- Nombre de points de vente,
- Qualité en valeur ajouté.

D) Si vous ne possédez pas d'imprimante

Remplacez les lignes de programmation 6100 à 6400 par:

```

6100 GOTO 6500
6500 PRINT" ** RESULTATS **"
6520 FOR I=0 TO N-1
6530 PRINT" Societe ";A$(I)
6540 PRINT" Prix de vente:";A(I+4*N);";F"
6550 Z=100*INT((A(I+5*N)*(1.5+RND(0)))*.0
1)
6555 IF INKEY$="" THEN 6555
6560 PRINT"Budget approx. marketing-pub"
;Z;"F"
6565 IF INKEY$="" THEN 6565
6570 PRINT"Nombre de points de ventes:";
A(I+6*N)
6575 IF INKEY$="" THEN 6575
6580 PRINT"Qualite en valeur ajoutee:";A
(I+7*N)
6585 IF INKEY$="" THEN 6585
6600 NEXT I
6700 GOTO 5080

```

E) Le vainqueur

Le gagnant est celui qui en fin de partie (soit au bout de six ans ou six périodes) possède le capital le plus élevé dans l'une de ses sociétés.

Programme:

```

5000 PRINT" * ETUDE DE MARCHE *"
5010 DIM A(100)
5020 INPUT"Nombre de societes";N
5030 FOR I=0 TO N-1
5040 PRINT"Nom de la societe numero ";I+
1;
5050 INPUT A$(I)
5060 A(I+N)=60000
5070 NEXT I
5080 FOR I=0 TO N-1

```

```

5090 A=A(I+N)
5100 PRINT "Societe ";A$(I)
5110 F=15000+1000*(INT(A(I+8*N)/499))
5115 IF INKEY$="" THEN 5115
5120 PRINT "Frais fixes=";F
5130 A=A-F
5140 IF A<0 THEN PRINT " * FAILLITE *** * ";A
(I+N)=-2:GOTO 5260
5150 K=K+1
5160 IF K=26 THEN 5120
5180 PRINT "Reste:";A;"F - Stock:";U;
5190 INPUT "- Vous fabriquez(10F/unite)";
Y
5195 A(I+8*N)=Y
5200 A(I+3*N)=A(I+3*N)+Y
5210 A=A-Y*10
5220 X=X+A(I+3*N)
5230 PRINT "Reste :";A
5240 INPUT "Supplement qualite/Cout fabrication";Q
5245 A(I+7*N)=Q
5250 A=A-A(I+3*N)*Q
5260 PRINT "Reste:";A
5400 INPUT "Nombre de points de ventes (Cout:100F)";P
5420 A=A-P*100
5430 PRINT "Reste:";A
5440 M=M*.2
5500 INPUT "Budget marketing-publicite";Z
5510 M=M+Z
5520 A=A-Z
5530 PRINT "Reste:";A
5540 INPUT "Prix de vente public";U
5550 A(I+4*N)=U
5560 K=(M^2)/10^8+1
5570 IF M>17000 THEN K=3.89+M/10^5
5580 L=(Q+1)^.3-.5

```

```


5590 IF U>100 THEN R=0:GOTO 5650
5600 IF U>39.9 THEN R=.28/(U-39):GOTO 56
50
5620 R=(1600-U^2)^.5/10
5650 T=40*P^.5*K*L*R*4
5660 IF T>8000 THEN T=7200+T*.1
5670 A(I+2*N)=T
5680 IF A>=0 THEN 5750
5700 PRINT "Decouvert a 20% de ";-A;
5710 PRINT " Soit : ";-A*.2;"F d'interet"
5720 A=A*1.2
5750 A(I+N)=A
5760 IF INKEY$="" THEN 5760
5763 A(I+5*N)=Z
5765 PRINT
5768 PRINT
5770 A(I+6*N)=P
5780 NEXT I
5800 X=ABS(1.1-X/100000)
5810 FOR I=0 TO N-1
5820 FOR J=0 TO N-1
5830 IF A(I+2*N)>A(J+2*N) THEN A(J+2*N)=
A(J+2*N)*X
5840 NEXT J
5850 NEXT I
5900 FOR I=0 TO N-1
5910 IF A(I+N)=-2 THEN 6050
5920 PRINT "Societe ";A$(I)
5925 IF INKEY$="" THEN 5325
5930 U=A(I+3*N)
5940 IF A(I+2*N)>U THEN C=U:GOTO 5960
5950 C=INT(A(I+2*N))
5960 PRINT "Ventes : ";C;"unites a ";A(I+4*N);"F"
5965 IF INKEY$="" THEN 5965
5970 D=A(I+4*N)*C
5980 A(I+N)=A(I+N)+D

```

```

5990 U=U-C
6000 PRINT"Reste stock:";U;"Cout 4F/unit
e"
6010 A(N+I)=A(N+I)-U*4
6020 PRINT" Caisse:";A(I+N)
6030 A(I+3*N)=U
6040 IF INKEY$="" THEN 6040
6042 PRINT
6044 PRINT
6050 NEXT I
6060 INPUT"fait-on une pause (O/N)";ZX$
6070 IF ZX$="O" THEN STOP
6080 IF ZX$="N" THEN 6100
6090 GOTO 6060
6100 ON ERROR GOTO 5080
6110 LPRINT[3,3]"      *** RESULTATS ***"
:
6120 FOR I=0 TO N-1
6130 LPRINT[2,2]" Societe ";A$(I)
6140 LPRINT[2,0]"Prix de vente:";A(I+4*N)
);"F"
6150 Z=100*INT((A(I+5*N)*(1.5+RND(0)))*.0
1)
6160 LPRINT[2,0]"Budget approx. marketin
g-pub";Z;"F"
6170 LPRINT[2,0]"Nombre de points de ven
tes:";A(I+6*N)
6180 LPRINT[2,0]"Qualite en valeur ajout
ee:";A(I+7*N)
6300 NEXT I
6400 GOTO 5080

```



Explications :

- 5010 On définit 100 mémoires numérotables : A(0), A(1), A(2), ..., A(98), A(99).
- 5020 N: Nombre de sociétés.
- 5030 I: Numéro de société.
- 5050 A\$(I): Nom de la société numéro I.
- 5060 A ou A(I+N): Capital ou reste.
- 5195 Y ou A(I+8×N): Quantité produite.
- 5200 U ou A(I+3×N): Quantité totale.
- 5245 Q ou A(I+7×N): Supplément qualité.
- 5400 P ou A(I+6×N): Nombre de points de vente.
- 5440 20 % sur l'avoir publicitaire de l'année précédente.
- 5540 V ou A(I+4×N): Prix de vente.
- 5560 K: coefficient publicité.
- 5580 L: coefficient qualité.

- 5620 R : coefficient prix.
- 5650 T : coefficient compétitivité (ou $A(I+2 \times N)$) ; ce coefficient correspond également à la quantité susceptible d'être acheté sans tenir compte de la concurrence.
- 5763 Z ou $A(I+5 \times N)$: Avoir publicitaire.
- 5830 Jeu de la concurrence selon la compétitivité où chaque entreprise est confrontée aux autres.
- 5940 C : quantités vendues.
- 5970 D : chiffre d'affaire.
- 6100 Si l'imprimante n'est pas sous tension le programme reprend sans donner les résultats généraux.
- 6110 LPRINT[3,3] signifie : imprimer en taille 3, en rouge (3) ce qui est entre guillemets.
- 6400 Retour à une nouvelle période ou année.

5

Jeux d'adresse

FIREFOX

Un jeu de réflexes et d'intuition qui vous passionnera lorsque vous le maîtriserez bien.

Le firefox est un vaisseau spatial. Vous en contrôlez la vitesse. Vous devez le faire avancer à l'intérieur d'un tracé de montagnes qui apparaît selon une loi aléatoire.

Vos points sont comptabilisés selon la vitesse de votre vaisseau à chaque pulsion du trajet signalée par deux Beep.

1 ^{re} vitesse :	1 point
2 ^e vitesse :	3 points
3 ^e vitesse :	8 points
4 ^e vitesse :	40 points (Très rapide)

Vous irez:


- Plus vite en pressant la touche **[2]**.
- Mois vite en pressant la touche **[1]**.
- A droite en pressant la touche **[^]**.
- A gauche en pressant la touche **[_]**.

Lors d'un resserrement du tracé, pensez à décélérer pour éviter de vous écraser sur une montagne.

Attention, n'appuyez qu'une seule fois à chaque pulsion car le programme mémorise les touches pressées les unes à la suite des autres.

A vous de jouer !

Pour rejouer, pressez la touche d'espacement; une musique vous donnera le départ.



Programme:

```
2000 PRINT" FIREFOX"
2002 FONT$(225)="136,248,248,112,112,32,
32,32"
2004 FONT$(149)="252,252,4,252,128,252,4
,252"
2005 GOSUB 3900
2010 Q=7:W=2:E=2:R=1000:T=0
2020 IF C$="1" THEN R=R+400
2030 IF C$="2" THEN R=R-400
2040 IF R<0 THEN R=0
2050 T=T+INT(2000/(R+50))
2090 BEEP 20,1
2100 I=0
2110 IF I>=Q THEN 2200
```

```

2120 PRINT" ";
2140 I=I+1 :GOTO 2110
2200 PRINT"█"; ← _____ [GRPH | V | GRPH]
2240 I=0
2250 IF I>=W THEN 2300
2260 PRINT" ";
2270 I=I+1 :GOTO 2250
2300 PRINT"▼"; ← _____ [GRPH | Z | GRPH]
2340 I=0
2350 IF I>=E THEN 2400
2360 PRINT" ";
2370 I=I+1 :GOTO 2350
2400 PRINT"█"; ← _____ [GRPH | V | GRPH]
2500 A=RND(0)
2510 IF A>.4 THEN Q=Q+1 :W=W-1
2520 IF A<=.4 THEN Q=Q-1 :W=W+1
2600 B=RND(1)
2610 IF B>.4 THEN E=E-1
2620 IF B<=.4 THEN E=E+1
2780 IF E+W+Q>16 AND Q>11 THEN Q=Q-1 :W=W
+1
2790 IF E+W+Q>16 THEN E=E-1
2795 IF Q<0 THEN Q=Q+1 :W=W-1
2800 IF E+W<2 AND E<1 THEN E=E+1
2810 IF E+W<2 THEN W=W+1 :Q=Q-1
2812 BEEP 25,1
2815 FOR I=0 TO R:NEXT I
2820 C$=INKEY$
2830 IF C$="-" THEN W=W-1 :E=E+1
2840 IF C$="^" THEN W=W+1 :E=E-1
2850 IF W<0 THEN 3000
2860 IF E<0 THEN 3200
2900 GOTO 2020
3000 FOR I=1 TO Q
3020 PRINT" ";
3040 NEXT I
3100 PRINT"*";
3120 FOR I=1 TO E

```


```

3140 PRINT" ";
3150 NEXT I
3170 PRINT ""
3180 GOTO 3500
3200 FOR I=1 TO Q
3220 PRINT" ";
3240 NEXT I
3260 PRINT "";
3270 FOR I=1 TO W
3280 PRINT" ";
3290 NEXT I
3300 PRINT "*"
3310 GOTO 3500
3500 PRINT "CRAACK! Score:";T
3600 GOSUB 3900
3650 IF INKEY$="" THEN 2000
3700 GOTO 3650
3900 BEEP 30,3:BEEP 28,3:BEEP 27,5
3910 BEEP 30,3:BEEP 35,5:BEEP 30,3:BEEP
35,9
3920 RETURN

```

Explications:

2002 On définit un caractère pour le dessin du vaisseau. On choisit le caractère 225 qui correspond aux touches **[GRPH]** puis **[Z]**. Pour cela on utilise le programme GRAPHO :



1 0 0 0 1 0	136
1 1 1 1 1 0	248
1 1 1 1 1 0	248
0 1 1 1 0 0	112
0 1 1 1 0 0	112
0 0 1 0 0 0	32
0 0 1 0 0 0	32
0 0 1 0 0 0	32

- 2004 On définit de la même façon les montagnes au caractère 149 qui correspond aux touches GRPH puis V.
- 2005 Le sous-programme qui se trouve ligne 3900 donne la musique de départ.
- 2010 On initialise les variables:
Q: Distance entre le bord gauche du tableau et la montagne de gauche. Au départ il y a sept espaces.
W: Distance entre la montagne de gauche et le vaisseau.
E: Distance entre le vaisseau et la montagne de droite.
R: Temps d'attente.
T: Score.
- 2020 C\$: Touche pressée définie en 2820.
Ici on règle le temps d'attente R.
- 2050 Cette fonction permet de donner 1 point pour la 1^{re} vitesse, 3 points pour la 2^e vitesse, 8 points pour la 3^e vitesse et 40 points pour la 4^e vitesse, qui s'ajoutent au score T.
- 2090 Première demi-pulsion (son).
- 2100 à 2140 On imprime Q espaces.
- 2200 On imprime la première montagne: pressez GRPH puis V pour le caractère entre guillemets.
- 2500 A: Déplacement au hasard de la montagne de gauche.
- 2600 B: Déplacement au hasard de la montagne de droite.
- 2780 à 2795 On évite de sortir de l'écran.
- 2800 à 2810 On assure deux espaces libres minimum entre les montagnes.
- 2812 Deuxième demi-pulsion (son).
- 2850 à 2860 Test de la collision.
- 2900 Retour pour une nouvelle pulsion.
- 3000 à 3180 Imprime une collision du côté gauche.
- 3200 à 3310 Imprime une collision du côté droit.
- 3650 Pour rejouer on presse la touche d'espacement.
- 3900 à 3920 Sous-programme musical.

PACMAN

Voici un jeu très divertissant parfaitement adapté au XO7.

Un petit bonhomme que vous contrôlez dans ses déplacements est poursuivi par un monstre plus rapide que lui.

Cependant, le bonhomme peut sortir d'un côté de l'écran et réapparaître de l'autre côté pour se protéger du monstre. Le monstre lui ne connaît pas cette possibilité. A droite s'affiche votre score.

A chaque battement de cœur du petit bonhomme vous gagnez 10 points. A chaque prime que vous attrapez (représentée par \$) vous gagnez 1000 points.

Le but du jeu est donc d'attraper un maximum de primes sans se faire manger. Lorsque vous vous êtes fait manger trois fois, la partie s'arrête.

Pour aller à droite, pressez la touche].

Pour aller à gauche, pressez la touche :.

Pour aller vers le haut, pressez la touche [.

Pour aller vers le bas, pressez la touche ?.

Il est inutile de presser plusieurs fois une même touche puisque la touche donne une direction au petit bonhomme, qui sera suivie tant que vous n'aurez pas pressé une autre touche.

Les primes apparaîtront au hasard sur l'écran mais ne resteront pas longtemps à leur place. Vous devez donc vous dépêcher de les saisir.

Programme:

```
50 PRINT" ** PACMAN **"  
100 FONT$(224)="0,112,216,124,28,120,240  
,0"  
120 FONT$(225)="0,56,108,248,224,120,60,  
0"  
140 FONT$(231)="0,160,112,40,32,88,64,0"  
160 FONT$(233)="0,20,56,80,16,104,8,0"
```

```
180 FONT$(229)="248,248,248,248,248,248,  
248,248"  
300 BEEP 20,4:BEEP 18,4:BEEP 17,7  
310 BEEP 25,6:BEEP 25,6:BEEP 20,5  
320 BEEP 27,8:BEEP 24,5:BEEP 25,13  
500 ' Déplacement curseur.  
530 C=30  
540 CLS  
600 Z$=INKEY$  
610 BEEP 27,1  
620 T=T+10  
630 IF INT(T/200)=T/200 THEN H=INT(RND(1  
)*71)  
650 IF Z$="]" THEN D=1  
660 IF Z$=":" THEN D=2  
670 IF Z$="[" THEN D=3  
680 IF Z$=? THEN D=4  
690 IF Z$="0" THEN D=5  
700 IF D=1 THEN C=C+1  
710 IF D=2 THEN C=C-1  
720 IF D=4 THEN C=C+20  
730 IF D=3 THEN C=C-20  
740 IF C=69 AND D=1 THEN C=0  
750 IF C<-9 THEN C=C+80  
760 IF C<-1 AND C>-10 THEN C=C+60  
770 IF C>79 THEN C=C-80  
780 IF C>68 AND C<80 THEN C=C-60  
820 IF C=-1 AND D=2 THEN C=70  
900 ' Monstres  
920 IF AW-CW<0 THEN A=A+1  
925 IF AW-CW>0 THEN A=A-1  
930 IF AQ<>CQ THEN A=A+20  
935 IF RND(0)<.1 THEN A=A+1  
940 IF RND(1)<.1 THEN A=A+20  
945 IF A>79 THEN A=A-80  
950 IF A>68 AND A<80 THEN A=A-60  
960 IF A=-1 THEN A=70
```

```

1000 IF FW-CW>0 THEN F=F-1
1010 IF FQ<>CQ THEN F=F+20
1020 IF FW-CW<0 THEN F=F+1
1100 BEEP 20,1
2000 ' IMPRIME
2020 LOCATE CW,CQ
2040 PRINT" ";
2100 CQ=INT(C/20)
2120 CW=C-CQ*20
2130 IF CQ<0 THEN CQ=0
2140 LOCATE CW,CQ
2145 IF D=2 THEN PRINT"¤" ←
2150 IF D=2 THEN 2200
2160 PRINT"¤";
2200 LOCATE HW,HQ
2210 PRINT" ";
2220 HQ=INT(H/20)
2230 HW=H-HQ*20
2240 LOCATE HW,HQ
2250 PRINT"$";
2300 LOCATE 10,3
2310 PRINT"¤T";
2400 LOCATE AW,AQ
2410 PRINT" ";
2420 AQ=INT(A/20)
2430 AW=A-AQ*20
2440 LOCATE AW,AQ
2450 IF AW-CW>0 THEN PRINT"¤";
2460 IF AW-CW>0 THEN 2500
2470 PRINT"¤";
2500 IF A=C THEN 4000
2600 IF H<>C THEN 600
2610 T=T+1000
2620 BEEP 30,5:BEEP 23,3:BEEP 21,2:BEEP
30,8
2900 GOTO 600
4000 LOCATE 5,1

```

Pour obtenir les caractères graphiques reportez-vous à la rubrique qui se trouve après les explications.

```
4100 PRINT"SLURP ¶";  
4150 BEEP 26,8:BEEP 21,3:BEEP 20,8  
4200 U=U+1  
4210 IF U<>3 THEN 500  
4220 PRINT  
4230 PRINT" ** GAME OVER **";  
4240 IF INKEY$="" THEN RUN  
4250 GOTO 4240
```


Explications :

- 100 Création du caractère "monstre gauche" accessible avec **GRPH 7**.
- 120 Création du caractère "monstre droite" accessible avec **GRPH Z**.
- 140 Création du caractère "bonhomme gauche" accessible avec **GRPH @**.
- 160 Création du caractère "bonhomme droite" accessible avec **GRPH 1**.
- 180 Création du caractère "Trait séparation" accessible avec **GRPH J**.
- 300 à 320 Musique de départ.
- 530 Curseur du bonhomme placé initialement à la case 30.
- 630 Toutes les 20 pulsions, la prime change de place.
- 650 à 730 Mouvement du curseur bonhomme.
- 740 à 820 Gestion du curseur bonhomme aux limites de l'écran.
- 920 à 930 Poursuite du bonhomme par le monstre.
- 935 à 1020 Gestion des sauts et mouvements du monstre.
- 2020 à 2040 On efface la case où se trouvait auparavant le coureur.
- 2100 à 2160 On imprime le coureur.
- 2200 à 2210 On efface la case où se trouvait avant la prime.
- 2220 à 2250 Et on imprime la nouvelle prime.
- 2300 à 2310 On imprime le score.

2400 à 2410 On efface la case où se trouvait avant le monstre.
2420 à 2470 Et on imprime le monstre.
2500 Testons si le monstre se trouve sur la case du bonhomme.
2600 Testons si le bonhomme se trouve sur la prime.
4200 U nombre de bonhommes mangés.
4210 Au bout de trois bonhommes mangés: fin du jeu.
4240 Presser la touche d'espacement pour recommencer.

Caractères graphiques:

2145 Caractère obtenu en faisant	GRPH	@	GRPH	.
2160 Caractère obtenu en faisant	GRPH	1	GRPH	.
2310 Caractère obtenu en faisant	GRPH	J	GRPH	.
2450 Caractère obtenu en faisant	GRPH	Z	GRPH	.
2470 Caractère obtenu en faisant	GRPH	7	GRPH	.



Exemple :

RUN

(musique de départ)

Pressez la touche  vous allez à droite.

Attendez la sortie de l'écran. Vous apparaîtrez au côté gauche de l'écran.

Pour aller à gauche pressez .

Pour aller en haut pressez .

Pour aller en bas pressez .

Le jeu s'achève quand vous vous êtes fait manger trois fois.

**** GAME OVER ****

Pour rejouer pressez la touche d'espacement.

6

Jeux de réflexion

Le pendu

D'une conception très sympathique, ce programme vous fera passer de bons moments.

Une musique annonce le départ.

Apparaissent sur l'écran des traits qui représentent les chiffres que vous aurez à trouver pour reconstituer le nombre cherché. Ce nombre sera obligatoirement le carré d'un nombre entier (ce qui vous orientera dans vos recherches).

Chaque fois que vous trouverez un chiffre composant le nombre en question, ce chiffre remplacera le trait à son emplacement.

Lorsque vous faites des erreurs, le pendu se dessine trait par trait; au bout de huit erreurs, vous êtes pendu et le nombre cherché s'affiche.

Attention ! Un chiffre contenu plusieurs fois dans le nombre cherché sera à trouver autant de fois.

Programme:

```
100 PRINT" ** LE PENDU **"
150 GOSUB6000
200 DIM A(10):DIM B(10)
300 A=INT(RND(0)*683)+317
400 B=A^2
500 FOR I=1 TO 6
600 A(I)=INT(B/10^(6-I))
700 B=B-A(I)*10^(6-I)
800 NEXTI
900 PRINT"Trouvez le carre:"
950 PRINT
1000 PRINT" - - - - - ";
1100 D$=INKEY$
1200 IF D$="" THEN 1100
1300 E=VAL(D$)
1400 FOR I=1 TO 6
1500 IF A(I)<>E OR B(I)=1 THEN 2000
1600 LOCATE2*I-2,3
1700 PRINTE;
1800 B(I)=1
1900 GOTO 1100
2000 NEXTI
2100 J=J+1
2200 ON J GOTO 4000,3800,3600,3400,3200,
3000,2800,2600
2600 LOCATE0,1
2650 PRINT" ** PENDU !! **"
2700 PRINT"C'etait:";A^2;
2730 GOSUB 6000
2750 END
2800 LINE(102,31)-(119,31)
3000 LINE(110,10)-(110,31)
3200 LINE(102,10)-(110,10)
3400 LINE(102,20)-(102,10)
3600 LINE(97,25)-(103,25)
```

```
3800 LINE(100,22)-(100,30)
4000 CIRCLE(100,20),2
5000 GOTO 1100
6000 BEEP 13,2:BEEP 16,3:BEEP 17,6
6100 BEEP 20,5:BEEP 18,5:BEEP 15,9
7000 RETURN
```

Explications :

- 200 On crée des mémoires numérotables.
- 300 On prend un nombre au hasard, mais en sorte que son carré soit composé de six chiffres.
- 600 On décompose dans les mémoires A(1), A(2), ..., A(6) les chiffres.
- 1300 E prend la valeur de la touche pressée.
- 1500 On vérifie si cette touche coïncide avec l'un des chiffres du nombre cherché.
- 1600 On positionne le curseur à l'emplacement du chiffre trouvé.
- 1800 Lorsque ce chiffre est imprimé, la mémoire B(I) correspondant à l'emplacement prend la valeur 1 et permet de ne plus considérer cet emplacement.
- 2200 Le pendu se dessine dans l'ordre. Il est possible d'augmenter la difficulté en réduisant le nombre d'erreurs possibles. Pour cela il suffit d'omettre ici certains numéros de ligne comme 3600 ou 3200...

Exemple :

RUN

Pressez les chiffres les uns à la suite des autres.

Méli-mélo

Présentation :

Vous avez aimé le "Rubick's Cube", vous aimerez ce jeu de logique.

Ce programme contient sept types de casses-têtes différents.

Une série de chiffres ordonnée: "123456789" sera mélangée Z fois; Z sera défini par le joueur selon la difficulté désirée.

Ces mélanges suivront une règle (Choisie par le joueur), mais à chaque mélange de l'ordinateur, l'endroit où cette règle s'appliquera, sera déterminé par le hasard.

Le but du jeu, sera de démêler, de *remettre en ordre* la série "123456789" en un minimum de coups, à l'aide des mêmes règles. Il appartient au joueur, de chercher par la logique l'endroit où cette règle s'appliquera, pour remettre en ordre la série.


7 casses têtes différents, sept jeux en 1:

A) Mélanges en rotation

Dans ce type de mélange une rotation à l'intérieur d'une partie de la série s'effectue. Le premier chiffre de cette partie remplace le dernier, et tous les autres chiffres de cette partie se déplacent d'un cran vers la gauche.

1) Rotation à trois chiffres:

Voici le schéma des quatre opérations possibles pour cette rotation:




Ainsi, lorsque l'on mélange la série "123456789" avec l'opération 1, elle devient: "231456789".

En appliquant l'opération 2 sur la série obtenue, on obtient: "234516789".

Et ainsi de suite.

2) Rotation à quatre chiffres:

Voici le schéma des opérations possibles :



Exemple :

"539681472" devient par l'opération 3 : "539614782".

3) Rotation à cinq chiffres:

Exemple :

"539681472" devient par l'opération 3 :

"539814762" puis par l'opération 2 :

"598143762" par l'opération 4 :

"598124376" par l'opération 1 :

"981254376".

4) Rotation à six chiffres:

Exemple :

"736829154" devient par l'opération 2 :

"768291354" ensuite par l'opération 4 :

"768913542" par l'opération 1 :

"689137542" par l'opération 3 :

"681375492".


On constate que la difficulté augmente avec le nombre de chiffres choisi.

B) Mélanges en permutation

Dans ce type de mélange, on imaginera la partie de la série permutée comme dans un miroir. Ainsi le premier chiffre de la partie et le dernier s'interchangent, le deuxième et l'avant dernier, et ainsi de suite.

1) *La permutation à trois chiffres* n'existe pas; elle ne permettrait pas de mélanger tous les éléments de la série.

2) *La permutation à quatre chiffres*:



Exemple :

"197538264" devient par l'opération 1:

"579138264" puis par l'opération 3:

"579162834".

3) *La permutation à cinq chiffres*:

Exemple :

"167832495" devient par l'opération 3:

"167942385" puis par l'opération 4:

"167958324" par l'opération 2:

"185976324" par l'opération 1:

"795816324".

4) *La permutation à six chiffres*:

Exemple :

"579162834" devient par l'opération 1:

"261975834" puis par l'opération 2:

"285791634" puis par l'opération 3:

"283619754" enfin par l'opération 4:

"283457916".

Programme:

```
700 DIM A(50)
800 PRINT" ** MELI-MELO ***"
900 INPUT"Nombre de chiffres qui s'inter
changent (3,4,5 ou 6)";C
910 IF C=0 THEN 10800
920 IF C<2 OR C>6 THEN 900
940 INPUT"Melange en rotation (R) ou en
permutation (P)";P$
950 IF C=3 THEN 980
960 IF P$="R" THEN C=C+4
980 IF P$<>"P" AND P$<>"R" THEN 940
1000 INPUT "Nombre de melanges au hasard
";Z
1010 IF Z=0 THEN 10800
1020 IF Z<1 THEN 1000
1030 IF Z>50 THEN Z=50
1050 M=0
1100 FOR I=1 TO 9
1200 A(I)=I
1300 NEXT I
1400 FOR I=1 TO Z
1500 N=INT(RND(1)*4)+1
1600 ON C GOSUB 900,900,3000,4000,5000,6
000,900,8000,9000,10000
1700 NEXT I
1710 S=0
1720 FOR I=1 TO 9
1730 S=(S+A(I))*10
1740 NEXT I
1750 S=S/10
1760 IF S=123456789 THEN 1900
1770 M=M+1
1780 PRINTS;":coup";M
1850 INPUT N
1855 IF N=0 THEN 10800
1860 IF N>4 OR N<1 THEN 1850
```

```
1880 ON C GOSUB 900,900,3000,4000,5000,6
000,900,8000,9000,10000
1890 GOTO 1710
1900 IF M=0 THEN 1400
1910 IF C=3 THEN Z=2*Z
1920 IF M>Z THEN 1960
1930 PRINT" BRAVO ! Trouve en ";M;"coups."
"
1940 GOTO 900
1960 PRINT"PAS MAL !";M;"coups"
1970 GOTO 900
3000 N=N*2-1
3100 T=A(N)
3200 A(N)=A(N+1)
3300 A(N+1)=A(N+2)
3400 A(N+2)=T
3500 RETURN
4000 IF N>2 THEN N=N+2
4100 T=A(N)
4200 A(N)=A(N+3)
4300 A(N+3)=T
4400 T=A(N+1)
4500 A(N+1)=A(N+2)
4600 A(N+2)=T
4700 RETURN
5000 IF N>2 THEN N=N+1
5100 T=A(N)
5200 A(N)=A(N+4)
5300 A(N+4)=T
5400 T=A(N+1)
5500 A(N+1)=A(N+3)
5600 A(N+3)=T
5700 RETURN
6000 T=A(N)
6100 A(N)=A(N+5)
6200 A(N+5)=T
6300 T=A(N+4)
```

```

6400 A(N+4)=A(N+1)
6500 A(N+1)=T
6600 T=A(N+2)
6700 A(N+2)=A(N+3)
6800 A(N+3)=T
6900 RETURN
7000 IF N>2 THEN N=N+2
7100 T=A(N)
7200 A(N)=A(N+1)
7300 A(N+1)=A(N+2)
7400 A(N+2)=A(N+3)
7500 A(N+3)=T
7600 RETURN
7700 IF N>2 THEN N=N+1
7800 T=A(N)
7900 A(N)=A(N+1)
8000 A(N+1)=A(N+2)
8100 A(N+2)=A(N+3)
8200 A(N+3)=A(N+4)
8300 A(N+4)=T
8400 RETURN
8500 T=A(N)
8600 A(N)=A(N+1)
8700 A(N+1)=A(N+2)
8800 A(N+2)=A(N+3)
8900 A(N+3)=A(N+4)
9000 A(N+4)=T
9100 RETURN
9200 T=A(N)
9300 A(N)=A(N+1)
9400 A(N+1)=A(N+2)
9500 A(N+2)=A(N+3)
9600 A(N+3)=A(N+4)
9700 A(N+4)=T
9800 RETURN
9900 T=A(N)
10000 A(N)=A(N+1)
10100 A(N+1)=A(N+2)
10200 A(N+2)=A(N+3)
10300 A(N+3)=A(N+4)
10400 A(N+4)=A(N+5)
10500 A(N+5)=T
10600 RETURN
10700 END

```

Explications :

700 On définit 50 mémoires numérotables.

960 Lorsque l'on veut faire une rotation l'indicateur C du jeu désiré est incrémenté de quatre unités.

- 1600 On se dirige vers le sous-programme du jeu désiré.
1760 On vérifie si la série est obtenue par le joueur.
1910 à 1960 Notation selon la vitesse à laquelle a été trouvée la solution.
3000 à 10000 Sous-programmes pour les différents jeux.

Exemple :

Rotation ou permutation ? R

Nombre de mélanges ? 2

312456978 ? 1

231456978 ? 1

123456978 ? 4

123456897 ? 4


BRAVO ! 4 coups

Celui-ci, bien sûr est un exemple simple.

L'ignoble SCHNOEKLE

L'ignoble, le terrible, l'immonde SCHNOEKLE se cache dans une grille de 10×10 cases.

En suivant les indications données par l'ordinateur (Glacial, tiède, ...) vous devez trouver l'endroit exact où se trouve le Schnoekle, mais gare à vous si vous vous trompez ; car le Schnoekle ne fera de vous qu'une bouchée. Puis ne mettez pas trop de temps à le chercher car au bout d'un certain nombre de coups, le Schnoekle se réveille et saute !



Programme:

```

10 PRINT "L'IGNOBLE SCHNOEKLE"
20 INPUT "Ignominie du Schnoekle comprise
entre 0 et 9"; T
30 INPUT "Profondeur du sommeil du Schnoe
kle"; S
40 PRINT "LE SCHNOEKLE SAUTE"
50 I=0
60 A=INT(RND(0)*10)

```

```
70 B=INT(RND(1)*10)
80 E=ABS(A-C)
90 F=ABS(B-D)
100 I=I+1
110 IF I>=S THEN 40
120 PRINTC;D
130 INPUT"Horizontal";H
150 PRINTC+H;D
160 INPUT"Vertical";U
170 C=C+H
180 IF C>=0 THEN 220
190 PRINT"Hors tableau"
200 C=C-H
210 GOTO120
220 IF C>9 THEN190
230 D=D+U
240 IF D>=0 THEN 270
250 D=D-U
260 GOTO 190
270 IF D>9 THEN 250
280 GOTO 400
290 Z=INT(((E-1-ABS(A-C))/10+1))
300 Z=Z+INT(((F-1-ABS(B-D))/10+1))
310 IF Z<>0 THEN 340
320 PRINT"GLACIAL"
330 GOTO 80
340 IF Z<>1 THEN 370
350 PRINT"TIEDE"
360 GOTO 80
370 IF Z<>2 THEN 220
380 PRINT"BRULANT"
390 GOTO 80
400 IF A=C THEN 490
410 IF ABS(A-C)>T THEN 430
420 IF B-D=0 THEN 460
430 IF ABS(B-D)>T THEN 450
440 IF A-C=0 THEN 460
450 GOTO 290
```

```
460 PRINT "GLOUP !! slurp !! MALHEUREUX, L' ignoble Schnoekle n'a fait de toi";  
470 PRINT " qu'une bouchee !";  
480 GOTO 550  
490 IF B<>D THEN 410  
500 PRINT "BRAVO !! CHAMPION !";  
510 PRINT "TU AS TUE L'AFFREUX, L'IGNOBLE SCHNOEKLE";  
550 IF INKEY$="" THEN 550  
560 RUN
```

Explications :

- 60 A est la coordonnée horizontale ou abscisse de l'emplacement du Schnoekle.
- 70 B est la coordonnée verticale.
- 80 C: coordonnée horizontale du joueur.
E: distance horizontale entre le Schnoekle et le joueur.
- 90 D: coordonnée verticale du joueur.
F: distance verticale entre le Schnoekle et le joueur.
- 100 I: nombre de coups.
- 290 et 300: fonction permettant de trouver Z en sorte que Z soit un indice pour le joueur.
- 400 à 450: Différents tests permettant de déceler si le Schnoekle mange le joueur ou si le joueur tue le Schnoekle.

Exemple :

En début de jeu, vous vous trouvez toujours dans la case de départ (0, 0).

Vous pouvez définir d'ignominie du Schnoekle, c'est-à-dire le nombre de cases en vertical ou en horizontal autour du Schnoekle sur lesquelles il pourra sévir. Prenons l'exemple du dessin :

Supposons que le Schnoekle se trouve dans la case (2, 8) et que vous ayez choisi une ignominie égale à 3. Ainsi, si vous vous trouvez dans l'une des cases suivantes vous serez dévoré !

(2, 7) (2, 6) (2, 5)
(3, 8) (4, 8) (5, 8)
(2, 9)
(1, 8) (0, 8)

Mais si vous parvenez à découvrir que le Schnoekle se trouve dans la case (2, 8), alors vous lui sauterez dessus et le tuerez.

L'ignominie conseillée du Schnoekle est de 2. La profondeur du sommeil du Schnoekle (soit le nombre de coups au bout desquels le Schnoekle sautera) est conseillée à 9.

A chaque mouvement votre position est indiquée. Au départ votre position est indiquée (0, 0).

Si vous voulez avancer à la case (6, 7) (cf. figure), vous ferez:

Horizontal ? 6

Vertical ? 7

L'ordinateur vous donnera une indication (exemple TIEDE) puis votre nouvelle position.

Si vous voulez maintenant avancer à la case (8, 3) vous devez faire un mouvement en horizontal de deux cases puis reculer en vertical de 3 cases. Vous répondrez donc :

Horizontal ? 2

Vertical ? -3

Une nouvelle indication vous sera donnée. (exemple: glacial).

Et ainsi de suite...

Quel est le sens des indications?

BRULANT: Signifie que vous vous approchez du Schnoekle en horizontal ET en vertical.

TIEDE: Signifie que vous vous approchez SOIT en horizontal SOIT en vertical.

GLACIAL: Signifie que vous ne vous approchez NI en horizontal NI en vertical.

Donc, faites preuve de logique et vous trouverez le Schnoekle sans vous faire dévorer !



Jeux de société

Jeu de dames

Jeu de réflexion pure, votre adversaire aux dames est l'ordinateur. Vous serez surpris de "l'intelligence" de ses réactions.

Vous disposerez d'un jeu de dames sur lequel vous aurez numéroté toutes les cases blanches comme l'indique la figure. (Prenez bien soin à ce que les numéros de votre tableau correspondent à ceux de la figure). Faites attention, il n'y a pas de case n° 11.

	1		2		3		4		5
6		7		8		9		10	
	12		13		14		15		16
17		18		19		20		21	
	23		24		25		26		27
28		29		30		31		32	
	34		35		36		37		38
39		40		41		42		43	
	45		46		47		48		49
50		51		52		53		54	

Malheureusement le programme ne fait pas la DAME : le vainqueur sera le premier à mettre un pion au bout du tableau. En effet, un programme de dames complet prendrait trop de place et serait trop long à retaper. Toutefois à partir des explications du programme, il vous sera possible de le compléter en faisant la DAME et même en ajoutant des degrés de réflexion (ces possibilités ont été prévues).

Pour jouer:

- Choisissez le niveau de difficulté (1 à 3) en sachant qu'au niveau 3 le temps moyen de réflexion est de 15 secondes pour obtenir une réponse de l'ordinateur. Au niveau 1 ce temps est de 10 secondes. Bien entendu, au niveau 3 il est bien plus difficile de battre l'ordinateur.
- Disposez vos pions de la case 34 à 54 et ceux de l'ordinateur de la case 1 à 21.
- Vous avez la possibilité de commencer mais également de permettre à l'ordinateur de démarrer en faisant:
Réponse à : "de la case ?" Taper: "36"
Réponse à : "à la case ?" Taper: "36".

- Vous pouvez "manger" en arrière et au cas où il y aurait plusieurs pions à manger, veuillez décomposer votre mouvement. L'ordinateur, lui, ne décomposera pas son mouvement, il indiquera uniquement la case de départ et celle d'arrivée. Il vous sera facile de voir les pions qui ont été mangés. Vous êtes obligé de manger quand vous en avez la possibilité.

Vous aurez toujours la possibilité de tricher en cas de coup dur, l'ordinateur ne vérifie pas.

Amusez-vous bien !

Programme :

```
100 PRINT" ** JEU DE DAMES **"
200 DIM A(80)
400 INPUT"Niveau (1 à 3)";P
500 IF P>3 THEN 400
600 P=(-1)*(P-4)
1100 FOR I=1 TO 21
1200 A(I)=1
1300 NEXT I
1400 FOR I=23 TO 32
1500 A(I)=3
1600 NEXT I
1700 FOR I=34 TO 54
1800 A(I)=8
1900 NEXT I
2000 A(1)=0:A(44)=0:A(61)=-6:A(62)=-5:A
(63)=5:A(64)=6
2100 INPUT"De la case";X
2200 INPUT"A la case";Y
2250 IF Y=0 THEN STOP:GOTO 2100
2300 IF INT(X+Y)<>X+Y THEN 2100
2500 A(X)=3:A(Y)=8
2600 IF ABS(X-Y)<8 THEN 3100
2700 A((X+Y)/2)=3
2800 INPUT"Remangez-vous (O/N)";R$
```

```
2900 IF R$="O" THEN 2100
3000 IF R$<>"N" THEN 2800
3100 I=1
3200 IF A(I)<>1 AND A(I)<>0 AND I>6 THEN
    Q=I-6:GOTO 3500
3300 I=I+1
3400 GOTO 3200
3500 B=0
3600 I=54
3700 IF A(I)=1 THEN GOSUB 5000
3800 IF B=7 THEN 20000
3850 I=I-1
3900 IF I<>0 THEN 3700
3950 B=0
4000 FOR L=P TO 3
4100 FOR W=49 TO Q STEP -1
4120 I=W
4140 IF INT((W-5)/11)=(W-5)/11 THEN I=W+
    5
4160 IF INT((W-10)/11)=(W-10)/11 THEN I=
    W+6
4200 FOR J=63 TO 64
4300 IF A(I)<>1 OR A(I+A(J))<>3 THEN 432
    0
4310 ON L GOSUB 8000,9000,10000
4320 IF B=7 THEN GOSUB 15000
4350 IF B=7 THEN 20000
4400 NEXT J
4500 NEXT W
4600 NEXT L
4700 GOTO 2100
5000 K=I
5100 FOR J=61 TO 64
5150 M=ABS(K+A(J)):N=ABS(K+2*A(J))
5200 IF A(M)=8 AND A(N)=3 THEN B=7:Y=N:X
    =I:A(M)=3:K=Y:J=60
5300 NEXT J
```

```

5400 RETURN
8000 IF I<13 THEN W=1:RETURN
8050 IF A(I+11)<>8 THEN RETURN
8100 IF A(I+2*A(J)-11)<>3 AND A(I-A(J)+2
2)<>8 THEN 8300
8200 GOTO 8500
8300 IF A(I+2*A(J))<>8 THEN 8400
8320 IF (A(I-5)+A(I-6))^2*(A(I-10)+A(I-1
2))>16 THEN 8500
8400 B=7:Y=I+A(J):X=I
8500 RETURN
9000 FOR K=61 TO 64
9100 IF A(I+A(J)+A(K))<>8 THEN NEXT K:X=
I:Y=I+A(J):B=7
9200 RETURN
10000 IF RND(1)>.7 THEN X=I:B=7:Y=I+A(J)
10100 IF I=0 THEN W=54
10200 RETURN
15000 IF A(I-A(J)+11)=1 AND A(I-2*A(J)+2
2)=8 THEN B=0
15100 RETURN
20000 PRINT "De la case";X;"à la case";Y
20100 A(X)=3:A(Y)=1
20200 GOTO 2100

```

Explications :

- 100 à 2000 emplacements des pions.
- 2100 à 3950 mouvements des pions.
- 4000 à 15100 stratégie (trois niveaux).
- 20000 à 20200 affichage.
- 200 On définit 80 mémoires numérotables :
A(1), A(2), A(3), ..., A(80).
- 1100 Les mémoires A(1) jusqu'à A(54) correspondent aux cases du tableau.

- 1200 Lorsque la mémoire $A(l) = 1$ cela signifie que dans la case numéro l il y a un pion appartenant à l'ordinateur.
- 1500 $A(l) = 3$: La case n° l est vide.
- 1800 $A(l) = 8$: La case n° l contient un pion du joueur.
- 2000 Les cases qui ne font pas partie du jeu sont égales à 0. Les mémoires $A(61)$, $A(62)$, $A(63)$, $A(64)$ correspondent aux différences qu'il y a entre une case quelconque du tableau et les quatre cases qui l'entourent.
- 2300 On vérifie qu'il n'y a pas d'erreur lorsque le joueur fait un mouvement.
- 2500 On vide la case X de départ du joueur et on met un pion à la case Y d'arrivée.
- 2600 Si l'on n'a pas mangé de pion il appartient maintenant à l'ordinateur de jouer.
- 2700 Mais si on a mangé, la case qui se trouve entre les deux cases du mouvement est vidée.
- 3100 Pour accélérer les réponses, on considérera que les pions susceptibles de bouger, sont les pions qui n'ont pas un pion de même couleur devant eux.
- 3600 à 3950 Sera considérée la plus grande case où il y aura un pion de l'ordinateur.
Voir lignes 5000 à 5400.
- 4310 On utilise les sous-programmes de stratégie dans l'ordre.
- 5000 à 5400 S'il y a possibilité de manger, on mange d'abord.
- 8000 à 8500 S'il y a possibilité de se placer sans danger afin de manger au prochain coup, l'ordinateur le fait.
- 9000 à 9200 Si l'ordinateur ne risque pas de se faire manger il avance en privilégiant les premières cases.
- 10000 à 10200 L'ordinateur bouge un pion quelconque.

Présentation du damier :

	1		2		3		4		5
6		7		8		9		10	
	12		13		14		15		16
17		18		19		20		21	
	23		24		25		26		27
28		29		30		31		32	
	34		35		36		37		38
39		40		41		42		43	
	45		46		47		48		49
50		51		52		53		54	

Exemple :

** Jeu de dames **

Niveau (1 à 3) ? 3

De la case ? 36

A la case ? 30

De la case 19 à la case 24

De la case ? 37

A la case ? 32

De la case 24 à la case 36 (ici votre pion qui se trouvait en case 30 a été mangé !)

De la case ? 42

A la case ? 30 (vous mangez)

Remangez-vous (O/N) ? N

De la case 21 à la case 27

Et ainsi de suite...

BACKGAMMON

Vous jouez contre l'ordinateur.

Vous serez étonné des performances de ce programme. En effet, des grands joueurs de Backgammon ont parfois été battus.

L'ordinateur ici réagit comme un véritable adversaire et dénonce les erreurs.

Cependant il serait bon de rappeler les règles du jeu pour ceux qui ne les connaissent pas, afin qu'ils puissent en profiter également.

Pour jouer vous vous munirez ou vous vous fabriquerez un tableau de Backgammon numéroté comme l'indique la figure.

Règles du jeu :

Dans un tableau (cf. figure) sont disposés vos pions "●" et ceux de l'ordinateur "○".


Le mouvement de vos pions s'effectue dans le sens décroissant des numéros des cases. Le mouvement de l'ordinateur s'effectue dans le sens croissant.


Le but du jeu

Le but de ce jeu est de placer tous vos pions dans les cases numérotées de 1 à 6 et au jet du dé retirer vos pions. Si vous avez fini de retirer tous vos pions avant l'ordinateur, vous avez gagné.

Déplacements

Les déplacements se font au jet du dé. Exemple : vous avez cinq pions dans la case 13 et vos dés affichent 4 et 3. Vous faites sauter un pion de la case 13 à la case 10. Lorsque vous faites un double (exemple 5 et 5) vous déplacez quatre fois vos pions de cinq cases.





Pour manger

Vous ne pouvez manger un pion de l'ordinateur que si vous tombez dessus et que s'il est seul dans sa case. Vous ne pouvez pas aller sur les cases qui contiennent deux pions ou plus de l'ordinateur.

Lorsqu'un pion est mangé il se place dans une case hypothétique (25 pour vous, 0 pour l'ordinateur), et vous ne pouvez jouer tant que votre pion n'est pas ressorti.

Programme :

```
100 PRINT" ** BACKGAMMON **"
200 DIM A(40)
300 A(1)=2:A(6)=-5:A(8)=-3:A(12)=5:A(13)
=-5:A(17)=3:A(19)=5:A(24)=-2
330 GOSUB 30000
350 GOSUB 20000
390 PRINT "Des :" ;A$;" ." ;AD
400 INPUT "JOUEZ" ;D
450 IF D=0 THEN 1900
460 IF D=-1 THEN GOSUB 30000:GOTO 390
500 E=INT(D)
600 F=100*(D-E)
650 IF F=0 AND A(E)<0 THEN A(E)=A(E)+1:G
OTO 1300
680 IF A(25)<0 AND E<>25 THEN 800
700 IF E>25 OR F>24 THEN 800
750 IF A(E)<0 AND A(F)<2 THEN 1000
800 PRINT"ERREUR !";
900 GOTO 400
1000 A(E)=A(E)+1
1100 IF A(F)=1 THEN A(F)=0:A(0)=A(0)+1
1200 A(F)=A(F)-1
1300 I=1
1400 IF A(I)<0 THEN 400
1450 I=I+1
1500 IF I<25 THEN 1400
```

```
1600 PRINT:PRINT" *** BRAVO ! *** VOUS  
AVEZ GAGNE !"  
1700 END  
1900 GOSUB 20000  
1950 PRINT AS;".";AD  
2000 INPUT "Mes des";L  
2020 IF L=-1 THEN GOSUB 30000:GOTO 1950  
2050 R=3:Q=3:UE=0  
2100 G=INT(L)  
2150 GA=0  
2200 H=10*(L-G)  
2300 IF G<7 AND H<7 AND H>0 THEN 2750  
2400 PRINT "ERREUR, ";  
2450 GOTO 2000  
2750 IF A(0)>0 THEN 19000  
2900 IF G>H THEN O=G:G=H:H=0  
3000 FOR M=1 TO 2  
3100 IF M=1 THEN P=G  
3150 IF Q=2 THEN M=2  
3200 IF M=2 THEN P=H  
3210 IF Q=1 THEN M=2:P=G  
3220 J=26  
3240 FOR I=1 TO 24  
3260 IF A(I)>0 THEN A(J)=I:J=J+1  
3280 NEXT I  
3290 IF A(26)>18 THEN 21000  
3300 FOR N=1 TO 7  
3400 FOR K=26 TO J-1  
3500 ON N GOTO 3600,4600,5600,6600,7600,  
8600,9600  
3600 IF A(A(K)+P)<>1 THEN 9990  
3650 IF A(K)+P>24 THEN 9990  
3700 PRINT A(K);",,";A(K)+P;":";  
3800 A(A(K))=A(A(K))-1:A(A(K)+P)=A(A(K)+  
P)+1:GA=GA+1  
3900 GOTO 18200  
4600 IF A(A(K)+P)=-1 AND A(K)+P<25 THEN  
A(25)=A(25)-1:A(A(K)+P)=0:GOTO 3700
```

```
4700 GOTO 9990
5600 IF M=1 AND ABS(A(A(K)+H))<2 AND H-G
=A(K+1)-A(K) THEN 5700
5650 GOTO 9990
5700 IF A(K)+H>24 THEN 9990
5800 PRINT A(K);",";A(K)+H;":";A(K+1);",
;A(K+1)+G;
5900 A(A(K))=A(A(K))-1:A(A(K)+H)=A(A(K)+
H)+2:GA=GA+1
5950 A(A(K+1))=A(A(K+1))-1
6000 M=2:GOTO 18200
6600 IF A(A(K))=1 AND A(A(K)+P)>-1 THEN
3650
6700 GOTO 9990
7600 IF A(A(K)+P)>-1 AND A(A(K))<>2 THEN
3650
7650 IF A(A(K)+P)>-1 AND G=H THEN 3650
7700 GOTO 9990
8600 IF A(A(K)+P)>-1 THEN 3650
9600 IF A(A(K)+P)>-1 THEN 3650
9990 NEXT K
18100 NEXT N
18200 NEXT M
18300 IF G=H AND R=3 THEN Q=3:R=5:GOTO 2
750
18400 GOTO 350
19000 IF A(G)<-1 AND A(H)<-1 THEN 19800
19100 IF A(G)<-1 THEN 19500
19200 PRINT "0,";G;
19250 IF A(G)=-1 THEN A(G)=0:A(25)=A(25)
-1
19300 A(0)=A(0)-1:A(G)=A(G)+1
19400 IF A(0)=0 THEN Q=2:GOTO 3000
19450 IF A(H)<-1 THEN 350
19500 IF A(H)<-1 THEN 350
19600 IF A(H)=-1 THEN A(H)=0:A(25)=A(25)
-1
```

```
19650 A(0)=A(0)-1:A(G)=A(G)+1
19700 IF A(0)=0 THEN Q=1:GOTO 3000
19750 GOTO 350
19800 PRINT"Bloque:";
19900 GOTO 18300
20000 'Jet de des
20100 AS=INT(RND(0)*6)+1
20200 AD=INT(RND(1)*6)+1
20400 RETURN
21000 IF H=0 THEN 22000
21100 GH=H:GA=GA+1
21200 IF GH<=25-A(26) THEN 21300
21250 A(A(26))=A(A(26))-1:PRINT A(26);",2
5:":GOTO 22000
21300 IF A(25-GH)<=0 THEN 21600
21400 A(25-GH)=A(25-GH)-1:PRINT 25-GH;",
25:":GOTO 22000
21600 A(A(26))=A(A(26))-1:A(A(26)+GH)=A(
A(26)+GH)+1
21700 PRINT A(26);".;"A(26)+GH;":";
22000 IF A(A(26))<=0 THEN A(26)=A(26)+1:
GOTO 22000
22030 IF A(26)>24 THEN PRINT:PRINT" **"
JE GAGNE !! **":END
22050 IF G=0 THEN 350
22100 IF GA=1 OR GA=3 THEN GA=GA+1:GH=G:
GOTO 21200
22200 IF G=H AND GA<4 THEN 21100
22500 GOTO 350
30000 PRINT" *** TABLEAU ***"
30050 LINE(0,23)-(119,23)
30100 FOR I=1 TO 12
30150 IF I=7 THEN LINE(0,23)-(119,23)
30200 IF I<10 THEN PRINT" ";
30300 PRINT I;
30400 X=A(I)
30500 GOSUB 33000
```

```
30520 QW=ABS(A(I)):QE=ABS(A(25-I))
30530 IF QW+QE>10 THEN 30600
30550 FOR QR=1 TO 11-QW-QE
30570 PRINT" ";
30580 NEXT QR
30600 X=A(25-I)
30700 GOSUB 33000
30800 PRINT25-I
30850 GH=100*VAL(INKEY$)
30900 FOR J=1 TO GH:NEXT J
31000 NEXT I
31100 LINE(0,23)-(119,23)
31200 IF A(0)<1 THEN 31500
31300 PRINT"Ordinateur:";
31350 X=A(0)
31400 GOSUB 33000
31450 PRINT
31500 IF A(25)>-1 THEN 32000
31600 PRINT"Joueur      :";
31650 X=A(25)
31700 GOSUB 33000
31800 PRINT
32000 RETURN
33000 IF X=0 THEN 33900
33100 IF X<0 THEN 33500
33200 FOR K=1 TO X
33300 PRINT"●";
33400 NEXT K
33450 GOTO 33900
33500 X=-X
33600 FOR K=1 TO X
33700 PRINT"○";
33800 NEXT K
33900 RETURN
```

Explications :

- 200 On définit 40 mémoires numérotables.
- 300 Emplacement des pions au départ. Vos pions sont négatifs, ceux de l'ordinateur positifs. Exemple : $A(13) = -5$, vous avez cinq pions dans la case 13.
- 330 Sous-programme affichage du tableau.
- 350 Sous-programme jet de dés.
- 650 Si la case d'arrivée F est la case 0 et la case de départ contient un de vos pions " $A(E) \leq 0$ " alors on retire un pion de la case de départ " $A(E) = A(E) + 1$ ".
- 700 à 900 Traitement des erreurs.
- 1450 à 1700 Test qui vérifiera si vous avez gagné.
- 2750 Si des pions de l'ordinateur ont été mangés, on les sortira d'abord.
- 2900 G est défini comme le plus petit dé et H comme le plus grand.
- 3220 à 3280 On décale la première case $A(J)$ contenant des pions de l'ordinateur et on numérote les suivantes.
- 3290 On teste si tous les pions de l'ordinateur se trouvent dans les six dernières cases.
- 3300 Il y a six niveaux de réflexion parcourus par l'ordinateur par ordre d'importance.
- 3400 Et on fait défiler toutes les cases contenant les pions de l'ordinateur à chaque niveau de réflexion.
- 3600 à 3900 Si il y a un pion de l'ordinateur seul, il sera protégé s'il peut l'être.
- 4600 à 4700 S'il y a un pion du joueur seul, il sera mangé.
- 5600 à 6000 S'il y a possibilité de bouger simultanément deux pions afin qu'ils se protègent mutuellement, ce sera fait.
- 6600 à 6700 On bougera d'abord les pions seuls.
- 7600 à 7700 On bougera ensuite les pions se trouvant dans une case où ils sont plus de deux.
- 8600 On bougera n'importe quel pion.
- 19000 à 19750 Sortie des pions de l'ordinateur.
- 19800 On est bloqué lorsque on ne peut pas les sortir.
- 20000 à 20300 Sous-programme du jet de dés.
- 21000 à 22500 Sortie des pions du dernier tableau.
- 30000 à 33900 On imprime le tableau.

Exemple:

RUN

Vous voyez apparaître le tableau qui défile sous vos yeux. Si le défilement est trop rapide vous avez la possibilité de le freiner en pressant la touche ... selon le temps d'attente désiré.

Dés 5 . 4
Jouez?

Vous voyez apparaître le jet des deux dés : 5 et 4. Vous n'êtes pas obligé de considérer les dés en question et vous pouvez utiliser des dés réels ; mais par souci de commodité une série aléatoire de deux dés apparaît.

Le droit de commencer la partie se joue à pile ou face, si vous gagnez, vous commencez.

Supposons que l'ordinateur commence !

Répondez Ø pour que l'ordinateur joue.

Jouez ?
4 .4
Mes dés?

(vous entrez les dés de l'ordinateur).

Mes dés ?
1,5 : 1,5 : 19,23 : 19,23
Dés : 5 . 2
Jouez?

Ici vous avez le mouvement de l'ordinateur qui déplace deux pions de la case 1 à la case 5 et deux pions de la case 19 à la case 23 car il a fait un double.

Vous avez la possibilité de consulter le tableau en faisant -1.

Jouez ?

tableau

Dés 5 . 2
Jouez?

Si vous déplacez un pion de la case 13 à 11 puis de 11 à 6, faites:

Jouez? 1 3 . 1 1

Jouez? 1 1 . Ø 6

Attention prenez bien garde de mettre un zéro devant le 6!!

Lorsque c'est à l'ordinateur de jouer, faites:

Jouez? Ø

Et ainsi de suite...

8

En cas d'erreur...

Tous les programmes de cet ouvrage ont été testés plusieurs fois et les listings sont une photocopie des listings de l'imprimante. Une erreur de listing serait donc tout à fait exceptionnelle.

Il est plus logique que vous ayez fait une erreur au moment du recopiage. Il est souvent facile de la déceler en listant la ligne de l'erreur et en la comparant.

Mais parfois, l'erreur n'apparaît pas tout de suite. Il faut chercher le type d'erreur (codifié par deux lettres), puis tester l'origine de l'erreur.

Une erreur fréquente de recopiage est de confondre O (zéro) avec O (la voyelle).

Une autre erreur est l'oubli d'inscrire une ligne du programme.



Tableau des codes de caractères

y \ x	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
0				0	@	P	'	p	♠	ì	√	-	タ	ミ	ô	ε
1			!	1	A	Q	a	q	♥	î	。	ア	チ	ム	ó	σ
2			"	2	B	R	b	r	♣	í	「	イ	ツ	メ	º	θ
3			#	3	C	S	c	s	♦	Ü	」	ウ	テ	モ	ÿ	x
4			\$	4	D	T	d	t	○	ü	,	エ	ト	ヤ	¢	λ
5			%	5	E	U	e	u	●	ù	。	オ	ナ	ユ	¢	μ
6			&	6	F	V	f	v	Ä	û	ヲ	カ	ニ	ヨ	Ñ	ρ
7	BELL		'	7	G	W	g	w	À	ú	τ	キ	ヌ	ラ	ñ	π
8			(8	H	X	h	x	ä	É	イ	ク	ネ	リ	Γ	τ
9)	9	I	Y	i	y	à	ë	ウ	ケ	ノ	ル	Σ	φ
10	L F		*	:	J	Z	j	z	â	è	エ	コ	ハ	レ	Π	χ
11	HOME		+	;	K	[k	{	á	ê	＊	サ	ヒ	ロ	Ω	ω
12	CLS	→	,	<	L	¥	l		å	é	ヤ	シ	フ	ワ	α	ν
13	CR	←	-	=	M]	m		¤	ö	ュ	ス	ヘ	ン	β	£
14		↑	.	>	N	^	n	~	İ	ö	ヨ	セ	ホ	ׇ	γ	¢
15		↓	/	?	O	_	o	i	ї	ò	ツ	ソ	マ	°	δ	÷

Peut être défini
par l'utilisateur

Peut être défini
par l'utilisateur

Achevé d'imprimer
par **Cid** éditions

Dépôt légal : avril 1984
N° d'Éditeur : 4086

Cet ouvrage vous plongera dans l'univers passionnant des jeux grâce au CANON X07.

Par le jeu vous découvrirez de nouvelles façons de programmer et de nombreuses astuces de programmation qui par la suite pourront vous être utiles. La complexité croissante des programmes vous permet de progresser à votre rythme et chaque ligne de programmation qui présente une difficulté est amplement expliquée.

Toutes sortes de jeux sont abordés, les jeux pratiques, graphiques, musicaux, mathématiques, économiques, jeux d'adresse, de réflexion et de société. Montrez vos dons de manager face à d'autres joueurs en sachant investir dans le Jeu d'entreprise. Ou alors battez votre propre record au Pacman puis essayez de battre l'ordinateur aux Dames ou au Backgammon.

Autant de défis qu'il vous faudra relever. Et en récompense assistez au Cinéma Musical qui vous surprendra.

Les jeux proposés seront pour vous une source abondante de divertissement. A vous de les découvrir !

1. **W**hat is the most important thing you can do to increase your sales?

2. **W**hat is the best way to increase sales?

3. **W**hat is the best way to increase sales?

4. **W**hat is the best way to increase sales?

5. **W**hat is the best way to increase sales?

6. **W**hat is the best way to increase sales?

7. **W**hat is the best way to increase sales?

8. **W**hat is the best way to increase sales?

9. **W**hat is the best way to increase sales?

10. **W**hat is the best way to increase sales?

